
                $PAGE(105)
                ;
                ; Pour visualiser ou imprimer correctement certains commentaires avec un éditeur
                ; de texte windauze, il est nécessaire d'utiliser une police à chasse fixe comme
                ; Fixedsys, Lucida Console, Courier, Courier New, DejaVu Sans Mono, Consolas...
                ;
                ; La Police "DejaVu Sans Mono" avec des caractères de Taille 7 et un Style Gras
                ; semble être la plus appropriée pour un listing au format A4.
                ;

                 ; Disassembled by:
                  ; DASMx object code disassembler
                  ; (c) Copyright 1996-2003   Conquest Consultants
                  ; Version 1.40 (Oct 18 2003)

                ;
                   ; File: RT86.bin

                ;
                   ; Size: 2048 bytes
                  ; Checksum: CC35
                   ; CRC-32: 224AE460

                ;
                ;

                   ; CPU: Intel 8049 (MCS-48 family)
                

                 ; ********************************
                 ; Affectation ports du µcontroleur
                 ; ********************************

                ;
                  ; DB0 (12) DI (3) EEROM
                  ; DB1 (13) SK (2) EEROM
                  ; DB2 (14) CS (1) EEROM
                   ; DB4 (15) Non connecté ; Voir en L027A 
                   ; DB5 (17) LED verte ; Allumage si 0
                   ; DB6 (18) LED jaune ; Allumage si 0
                   ; DB7 (19) TDA9045 (8) ; Sélection temporisation de détection du niveau du blanc

                ;
                   ; P10 (27) embase DIN (1) ; Liaison série half duplex à 4800 Bds
                  ; P11 (28) DO (4) EEROM
                    ; P12 (29) ; Lancement de l'auto-test si à 0 lors d'un reset
                    ; P13 (30) ; Colonne centrale du clavier
                    ; P14 (31) ; Colonne gauche du clavier
                    ; P15 (32) ; Colonne droite du clavier
                    ; P16 (33) : Démodulateur audio en service si 0
                    ; P17 (34) ; Activation commutation lente sur péritel si 0

                ;
                    ; P20 (21) ; 1ère rangée (rangée supérieure) du clavier
                    ; P21 (22) ; 2ème rangée du clavier
                    ; P22 (23) ; Dernière rangée (rangée inférieure) du clavier
                    ; P23 (24) ; 3ème rangée du clavier
                    ; P24 (35) ; Sélection du retard 0 si 0
                    ; P25 (36) ; Sélection du retard 1 si 0
                    ; P26 (37) ; Sélection du retard 2 si 1
                    ; P27 (38) ; Checksum de la ROM interne si à 0 lors d'un reset

                ;
                     ; T0 (1) ; Synchronisation trame, à 1 pendant l'impulsion
                     ; T1 (39) ; Synchronisation ligne, à 1 pendant l'impulsion
                     ; ; Cette impulsion incrémente le compteur d'évènements CNT
                    ; INT/ (6) ; Détection de niveau, à 1 si présence d'une ligne blanche

                ;
                 ; *********************
                 ; Octets en RAM interne
                 ; *********************

                ;
      0020 =         RAM020 EQU 20H ; Zone de 4 octets pour stocker la clef utilisée
      0026 =         RAM026 EQU 26H ; Sauvegarde résultat comparaison des CRC4
      0029 =         RAM029 EQU 29H ; Valeur/chiffre associé à la touche frappée
      002A =         RAM02A EQU 2AH ; Compteur de trames pour faire clignoter la LED verte
      002B =         RAM02B EQU 2BH ; Numéro de clef × 4 reçu dans la séquence
      002C =         RAM02C EQU 2CH ; Niveau d'audience reçu dans la séquence
      002F =         RAM02F EQU 2FH ; Contient adresse où recopier une des clefs
      0030 =         RAM030 EQU 30H ; Etat binaires des lignes, 2 × 12 octets

                     ; Entrée des états par la droite de chaque octet
                     ;      0 : 310 | 12
                     ;     +1 : 622 | 11
                     ;     +2 : 310 | 10
                     ;     +3 : 622 |  9
                     ;     +4 : 310 |  8
                     ;     +5 : 622 |  7  Ordre
                     ;     +6 : 310 |  6  temporel
                     ;     +7 : 622 |  5  inverse
                     ;     +8 : 310 |  4
                     ;     +9 : 622 |  3
                     ;    +10 : 310 |  2
                     ;    +11 : 622 |  1

      0048 =         RAM048 EQU 48H ; Six compteurs à rebours
      004E =         RAM04E EQU 4EH ; Compteur défauts centrage synchronisation trame
      004F =         RAM04F EQU 4FH ; Status de l'appareil

                     ; Bit 7 = synchronisé sur séquence simple/double
                     ; Bit 4 = 1 si brouillage des lignes

      0050 =         RAM050 EQU 50H ; Cumul certificats + états lignes 310
      0052 =         RAM052 EQU 52H ; Cumul certificats + états lignes 622
      005F =         RAM05F EQU 5FH ; Zone de stockage des chiffres entrés au clavier
   006A =         RAM06A EQU 6AH
      0070 =         RAM070 EQU 70H ; Table où sont stockées les 4 clefs désencryptées

                ;

                 ; *********************************
                 ; Touches non numériques du clavier
                 ; *********************************

                ;
      0010 =         ENT EQU 10H ; Valeur numérique de la touche "ENT"

      0020 =         MEMO EQU 20H ; Valeur numérique de la touche "MEMO"
                ;

                 ; *************************************
                 ; Valeurs correspondantes aux 3 retards
                 ; *************************************

                ;
      00AF =         R0000 EQU 0AFH ; Pas de retard, vidéo en direct
      009F =         R0902 EQU 9FH ; Retard de 902 ms
      00FF =         R1804 EQU 0FFH ; Retard de 1804 ms

                ;
                 ; *****************
                 ; Clefs prédéfinies
                 ; *****************

                ;
      29CB =         DEFAUT EQU 29CBH ; Clef par défaut à l'initialisation
      A72C =         SESAME EQU 0A72CH ; Clef pour le niveau d'audience 7

                ;
                 ; ************************
                 ; PROGRAMME du µControleur
                 ; ************************

                ;
      0000           ORG 0 ; Page 0

                ;
                L0000:

     0000 23FF      MOV A,#11111111B ; Initialisation des ports de P2 en entrée
   0002 3A        OUTL P2,A
      0003 0A        IN A,P2 ; Lecture P27
   0004 F210      JB7 L0010
      0006 E4DD      JMP L07DD ; Calcul du checksum si P27 à zéro

                ;
                 ; ********************************
                 ; Calcul du checksum sur la page 0
                 ; ********************************

                ;
                L0008:

      0008 FA        MOV A,R2 ; Adresse sur 8 bits dans la page
   0009 A3        MOVP A,@A
      000A F4ED      CALL L07ED ; Calcul de checksum sur 16 bits
   000C EA08      DJNZ R2,L0008
      000E 2400      JMP L0100 ; Poursuite du calcul sur la page 1

                ;
                 ; *********************************************************
                 ; Initialisation à la mise sous tension ou suite à un reset
                 ; *********************************************************

                ;
                L0010:

     0010 23F8      MOV A,#11111000B ; DI, CK et CS de l'EEROM forcés à zéro
   0012 02        OUTL BUS,A
      0013 B87F      MOV R0,#127 ; Dernière adresse en RAM

                L0015:
      0015 B000      MOV @R0,#0 ; Initialisation de la RAM et des registres
   0017 E815      DJNZ R0,L0015
      0019 BB1E      MOV R3,#30 ; Temporisation de 30 × ~20 ms

                L001B:
      001B D4AB      CALL L06AB ; Temporisation ~20 ms
   001D EB1B      DJNZ R3,L001B

                ;
                 ; **********************************************************************
                 ; Uniformisation des clefs de 26 bits à 32 bits
                 ;
                 ; Avant uniformisation
                 ; [<--- RAM020 -->|<-- RAM020+1 ->|<-- RAM020+2 ->|<-- RAM020+3 ->|
                 ; [3|3|2|2|2|2|2|2|2|2|2|2|1|1|1|1|1|1|1|1|1|1|0|0|0|0|0|0|0|0|0|0]
                 ; [1|0|9|8|7|6|5|4|3|2|1|0|9|8|7|6|5|4|3|2|1|0|9|8|7|6|5|4|3|2|1|0]
                 ; |25 <------- Clef intermédiaire  26 bits -------> 00| N°|<CRC 4>|
                 ;
                 ; 1er opération : annulation du quartet droit de RAM020+1
                 ; [<--- RAM020 -->|<-- RAM020+1 ->|<-- RAM020+2 ->|<-- RAM020+3 ->|
                 ; [3|3|2|2|2|2|2|2|2|2|2|2|×|×|×|×|1|1|1|1|1|1|0|0|0|0|0|0|0|0|0|0]
                 ; [1|0|9|8|7|6|5|4|3|2|1|0|0|0|0|0|5|4|3|2|1|0|9|8|7|6|5|4|3|2|1|0]
                 ;
                 ; 2ème opération : annihilation du CRC4 par quartet droit de RAM020+1
                 ; [<--- RAM020 -->|<-- RAM020+1 ->|<-- RAM020+2 ->|<-- RAM020+3 ->|
                 ; [3|3|2|2|2|2|2|2|2|2|2|2|×|×|×|×|1|1|1|1|1|1|0|0|0|0|0|0|1|1|1|1]
                 ; [1|0|9|8|7|6|5|4|3|2|1|0|0|0|0|0|5|4|3|2|1|0|9|8|7|6|5|4|9|8|7|6]
                 ;
                 ; 3ème opération : annulation du n° de clef (bits 05 et 04)
                 ; [<--- RAM020 -->|<-- RAM020+1 ->|<-- RAM020+2 ->|<-- RAM020+3 ->|
                 ; [3|3|2|2|2|2|2|2|2|2|2|2|×|×|×|×|1|1|1|1|1|1|0|0|0|0|×|×|1|1|1|1]
                 ; [1|0|9|8|7|6|5|4|3|2|1|0|0|0|0|0|5|4|3|2|1|0|9|8|7|6|0|0|9|8|7|6]
                 ;
                 ; ***********************************************************************

                ;
                L001F:

      001F D411      CALL L0611 ; Numéro d'appareil lu dans EEROM et stocké en RAM06A
      0021 BD04      MOV R5,#4 ; N° de registre de l'EEROM où est stocké la 1er clef
     0023 B82F      MOV R0,#RAM02F ; Contient adresse où recopier la clef désencryptée
     0025 B070      MOV @R0,#RAM070 ; Table où sont stockées les 4 clefs désencryptées

                L0027:

     0027 B820      MOV R0,#RAM020 ; Adresse où stocker la clef intermédiaire extraite de l'EEROM
      0029 BC02      MOV R4,#2 ; 2 mots de 16 bits à extraire
      002B D44C      CALL L064C ; Lecture dans l'EEROM
      002D 74D0      CALL L03D0 ; Désencryptage de la clef intermédiaire
   002F 27        CLR A
   0030 B921      MOV R1,#RAM020 + 1
      0032 31        XCHD A,@R1 ; Annulation du quartet droit de RAM020+1
   0033 19        INC R1
   0034 19        INC R1
      0035 31        XCHD A,@R1 ; Remplacement du CRC4 par le quartet droit de RAM020+1
      0036 F1        MOV A,@R1 ; 
     0037 53CF      ANL A,#11001111B ; Annulation du n° de clef (bits 05 et 04)
   0039 A1        MOV @R1,A
     003A B920      MOV R1,#RAM020 ; Adresse clef qui vient d'être uniformisée
     003C B82F      MOV R0,#RAM02F ; Adresse de destination stockée en RAM02F
   003E F0        MOV A,@R0
      003F A8        MOV R0,A ; Adresse destination copie dans R1
      0040 BA04      MOV R2,#4 ; Clef sur 4 octets
      0042 94F5      CALL L04F5 ; Recopie de la clef uniformisée
      0044 1D        INC R5 ; Registre EEROM suivant car R5 = dernier registre lu
     0045 B92F      MOV R1,#RAM02F ; Sauvegarde adresse recopie clef suivante
   0047 F1        MOV A,@R1
      0048 0304      ADD A,#4 ; Calcul adresse clef suivante
   004A A1        MOV @R1,A

      0010 =         Tsize SET 4 * 4 ; Table de 4 clefs sur 26 bits/4 octets
    004B D380      XRL A,#RAM070 + Tsize ; Test si toutes les clefs ont été uniformisées 
      004D 9627      JNZ L0027 ; Même opération sur clef suivante

                ;
                 ; *********************************************************************
                 ; Initialisation avant toute tentative de vérouillage sur synchro trame
                 ; *********************************************************************

                ;
                L004F:

     004F 89C0      ORL P1,#11000000B ; Désactivation commutation lente et démodulation audio
     0051 8820      ORL BUS,#00100000B ; DB5 à 1 : Extinction LED verte
     0053 B830      MOV R0,#RAM030 ; Adresse zone mémoire à initialiser
      0055 BA24      MOV R2,#36 ; Zone de 36 octets
      0057 54EF      CALL L02EF ; Initialisation des octets de cette zone

                ;
                 ; **********************************************************************
                 ; Tentative de vérouillage sur la synchro trame précédant la trame paire
                 ; **********************************************************************

                ;
                L0059:

   0059 2302      MOV A,#2
      005B 62        MOV T,A ; Initialisation du compteur de lignes
      005C 45        STRT CNT ; Démarrage du comptage des lignes

                L005D:
      005D 54E4      CALL L02E4 ; Scrutation du clavier
      005F 9663      JNZ L0063 ; Si rien ou touche autre que touche ENT
      0061 6408      JMP L0308 ; Entrée code périodique avec clignotement LED jaune

                ;
                L0063:

      0063 3663      JT0 L0063 ; Attente disparition synchro trame si présente
                L0065:

      0065 3681      JT0 L0081 ; Attente apparition d'une nouvelle synchro trame
      0067 4665      JNT1 L0065 ; Attente du top de synchro ligne suivant
     0069 99C7      ANL P1,#11000111B ; Colonnes clavier à zéro pour globaliser la détection
      006B 0A        IN A,P2 ; Lecture des 4 rangées du clavier
     006C 8938      ORL P1,#00111000B ; Sondage clavier terminé
   006E 37        CPL A
   006F 530F      ANL A,#00001111B
      0071 965D      JNZ L005D ; Scrutation du clavier si présence de touche
      0073 09        IN A,P1 ; Test de la liaison série sur P10/contact 1 de la DIN
   0074 127B      JB0 L007B
      0076 09        IN A,P1 ; Re-test de la liaison série
   0077 127B      JB0 L007B
       0079 6439      JMP L0339 ; Si un "break" est détecté sur cette liaison série

                ;
                L007B:

      007B 5265      JB2 L0065 ; Si port P12 non forcé à zéro
      007D F408      CALL L0708 ; Test clavier par frappe touches 1 à 9, 0, ENT et MEMO
      007F E43A      JMP L073A ; Test d'écriture sur registre 15 de l'EEROM

                ;
                L0081:

      0081 4681      JNT1 L0081 ; Attente apparition d'un top de synchro ligne
                L0083:

      0083 5683      JT1 L0083 ; Attente fin de ce top
   0085 42        MOV A,T
     0086 D33A      XRL A,#314 MOD 256 ; Test si ligne 314 (256 + 58)
      0088 9659      JNZ L0059 ; Rebouclage sur nouvelle tentative
   008A 2302      MOV A,#2
      008C 62        MOV T,A ; Rechargement du compteur ligne
      008D BF02      MOV R7,#2 ; Compteur limité aux 2 trames d'une image pendant synchronisation

                ;
                 ; ****************************************************************
                 ; Programme principal, reboucle sur lui même chaque nouvelle trame
                 ; ****************************************************************

                ;
                L008F:

      008F 09        IN A,P1 ; Test de la liaison série sur P10/contact 1 de la DIN
   0090 1297      JB0 L0097
      0092 09        IN A,P1 ; Re-test de la liaison série
   0093 1297      JB0 L0097

      0095 6439      JMP L0339 ; Si un "break" est détecté sur cette liaison
                ;
                L0097:

      0097 529D      JB2 L009D ; Si port P12 non forcé à zéro
      0099 F408      CALL L0708 ; Test clavier par frappe touches 1 à 9, 0, ENT et MEMO
      009B E43A      JMP L073A ; Test d'écriture sur registre 15 de l'EEROM

                ;
                L009D:

      009D 5410      CALL L0210 ; Défilement des lignes visibles + ligne 310/622
      009F BA05      MOV R2,#5 ; Nombre de lignes sur lesquelles on recherche la synchro trame
      00A1 FF        MOV A,R7 ; Numéro de trame
   00A2 12A5      JB0 L00A5
      00A4 1A        INC R2 ; Une ligne de plus si trame paire

                L00A5:
      00A5 2302      MOV A,#2 ; La synchro trame doit apparaitre ligne 2 ou 314

                L00A7:
      00A7 36B1      JT0 L00B1 ; Sortie boucle si apparitionn nouvelle synchro trame
      00A9 46A7      JNT1 L00A7 ; Attente apparition d'un top de synchro ligne

                L00AB:
      00AB 56AB      JT1 L00AB ; Attente fin de ce top
   00AD EAA7      DJNZ R2,L00A7
      00AF 2304      MOV A,#4 ; Ligne n° 4 ou 316 par défaut si synchro trame absente

                L00B1:
      00B1 46B1      JNT1 L00B1 ; Attente apparition du top de synchro ligne suivant

                L00B3:
      00B3 56B3      JT1 L00B3 ; Attente fin de ce top
   00B5 62        MOV T,A
      00B6 CF        DEC R7 ; Changement de trame donc décrément de son numéro
     00B7 B94E      MOV R1,#RAM04E ; Compteur de défaut synchro trame
   00B9 FA        MOV A,R2
      00BA D303      XRL A,#3 ; Centrée si elle apparait à la 3ème ligne 
      00BC 96C1      JNZ L00C1 ; Synchro trame non centrée sur ligne 2 ou 314
      00BE A1        MOV @R1,A ; Reset du compteur de défauts centrage trame
   00BF 04CB      JMP L00CB

                L00C1:
   00C1 FB        MOV A,R3
     00C2 537F      ANL A,#01111111B ; L'état de la dernière ligne 310/622 devient alors indécis
   00C4 AB        MOV R3,A
      00C5 11        INC @R1 ; Incrémentation du compteur de défaut
   00C6 F1        MOV A,@R1
   00C7 03FA      ADD A,#-6
      00C9 F64F      JC L004F ; Resynchronisation nécessaire si plus de 6 erreurs

                L00CB:
      00CB 54BB      CALL L02BB ; Cumul des états/bits de certification lignes 310/622
      00CD 54E4      CALL L02E4 ; Scrutation du clavier
      00CF 96D3      JNZ L00D3 ; Si rien ou touche autre que touche ENT
      00D1 6408      JMP L0308 ; Entrée code périodique avec clignotement LED jaune

                ;
                L00D3:

   00D3 3408      CALL L0108
   00D5 048F      JMP L008F

                ;
      0100           ORG 0100H ; Page 1

                ;
                 ; ********************************
                 ; Calcul du checksum sur la page 1
                 ; ********************************

                ;
                L0100:

      0100 FA        MOV A,R2 ; Adresse sur 8 bits dans la page
   0101 A3        MOVP A,@A
      0102 F4ED      CALL L07ED ; Calcul de checksum sur 16 bits
   0104 EA00      DJNZ R2,L0100
      0106 4408      JMP L0208 ; Poursuite du calcul sur la page 2

                ;
                L0108:

     0108 B84F      MOV R0,#RAM04F ; Status de l'appareil
   010A F0        MOV A,@R0
      010B F242      JB7 L0142 ; Appareil synchronisé

                ;
                 ; *******************************************************************************

                
                 ; Appareil NON synchronisé
                 ; Recherche des motifs sur les lignes 310
                  ; Séquence simple si motif élémentaire répété toutes les 3 lignes (011)
                  ; Séquence double si motif élémentaire répété toutes les 6 lignes (010.011)
                 ; Sortie
                  ; F1 à 0 si séquence simple
                  ; F1 à 1 si séquence double
                 ; *******************************************************************************

                
                 ;

      010D FF        MOV A,R7 ; Numéro de trame, le n° zéro est transitoire
      010E 9668      JNZ L0168 ; Traitement exclu si numéro de trame autre que zéro
      0110 BF02      MOV R7,#2 ; Compteur limité aux 2 trames d'une image pendant synchronisation
      0112 85        CLR F0 ; Indication simple séquence par défaut
   0113 B950      MOV R1,#RAM050
      0115 F1        MOV A,@R1 ; Validité des états des lignes 310 les plus vieilles
   0116 37        CPL A
      0117 9668      JNZ L0168 ; Tous ces états doivent être certifiés
   0119 19        INC R1
         011A F1        MOV A,@R1 ; Validité des états des lignes 310 les plus récentes
   011B 37        CPL A
      011C 9668      JNZ L0168 ; Tous ces états doivent être certifiés

     011E B952      MOV R1,#RAM052 ; 1er motif recherché sur 16 bits : 1.011.011.011.011.011
     0120 23B6      MOV A,#10110110B ; 1er octet du motif pour les bits les plus anciens
   0122 D1        XRL A,@R1
      0123 962B      JNZ L012B ; 1er motif non reconnu
   0125 19        INC R1
     0126 23DB      MOV A,#11011011B ; 2ème octet du motif pour les bits les plus récents
   0128 D1        XRL A,@R1
       0129 C639      JZ L0139 ; Motif de la séquence simple à 6 trames reconnu

                L012B:
   012B B952      MOV R1,#RAM052
       012D F1        MOV A,@R1 ; 2ème motif recherché sur 16 bits : 0.011.010.011.010.011
     012E D334      XRL A,#00110100B ; 1er octet du motif pour les bits les plus anciens
      0130 9668      JNZ L0168 ; 1er motif non reconnu
   0132 19        INC R1
   0133 F1        MOV A,@R1
     0134 D3D3      XRL A,#11010011B ; 2ème octet du motif pour les bits les plus récents
      0136 9668      JNZ L0168 ; Motif de la séquence double à 12 trames reconnu
      0138 95        CPL F0 ; Séquence double reconnue

                L0139:
      0139 34D9      CALL L01D9 ; Remplissage table états lignes 310/622 par une séquence double
     013B 997F      ANL P1,#01111111B ; Activation commutation lente sur péritel
     013D B84F      MOV R0,#RAM04F ; Status de l'appareil
     013F B080      MOV @R0,#1 SHL 7 ; Appareil déclaré synchronisé sur séquence
  0141 83        RET

                ;
                 ; ********************
                 ; Appareil synchronisé
                 ; ********************

                ;
                L0142:

      0142 FF        MOV A,R7 ; Numéro de trame, le n° 0 est transitoire avant n° 6/12
      0143 964D      JNZ L014D ; Si n° de trame autre que zéro
      0145 BF0C      MOV R7,#12 ; N° de trame de trame rechargé à 12
   0147 B64B      JF0 L014B
      0149 BF06      MOV R7,#6 ; N° de trame de trame rechargé à 6

                L014B:
      014B 8408      JMP L0408 ; Cas des trames n° 6 ou n° 12

                ;
                L014D:

   014D 07        DEC A
      014E C658      JZ L0158 ; Cas de la trame numéro 1 : L04A3
   0150 07        DEC A
      0151 C65A      JZ L015A ; Cas de la trame numéro 2
   0153 07        DEC A
      0154 9668      JNZ L0168 ; Pas de traitement spécifique pour les autres
      0156 A408      JMP L0508 ; Cas de la trame numéro 3

                L0158:
      0158 84A3      JMP L04A3 ; Cas de la trame numéro 1

                ;
                 ; ********************************************
                 ; Traitement spécifique en début de trame n° 2
                 ; ********************************************

                ;
                L015A:

      015A 3469      CALL L0169 ; Reconnaissance d'un motif sur lignes 310
      015C C668      JZ L0168 ; Motif sequence simple/double reconnu
     015E 89C0      ORL P1,#11000000B ; Désactivation commutation lente et démodulation audio
     0160 8820      ORL BUS,#00100000B ; Extinction LED verte 
      0162 BF02      MOV R7,#2 ; Compteur limité aux 2 trames d'une image pendant synchronisation
   0164 B84F      MOV R0,#RAM04F
      0166 B000      MOV @R0,#0 ; Désynchronisation, reset du status

                L0168:
  0168 83        RET

                ;
                 ; ************************************************************
                 ; Cumul des états des lignes 310 sur les 5 dernières séquences
                 ; ou uniquement sur la dernière si R3 = 00000001B 
                 ; ************************************************************

                ;
                L0169:

     0169 BB1F      MOV R3,#00011111B ; Cumul des 5 dernières séquences
                L016B:

     016B B850      MOV R0,#RAM050 ; Adresse octets à initialiser
   016D BA02      MOV R2,#2
      016F 54EF      CALL L02EF ; Initialisation de ces 2 octets
     0171 B851      MOV R0,#RAM050 + 1 ; Cumul des états

      0018 =         Tsize SET 12 * 2 ; Table de 12 lignes de 2 octets pour la séquence double
    0173 B948      MOV R1,#RAM030 + Tsize ; Sinon l'assembleur CYS8049 bogue sur #RAM030+(12×2)
      0175 BEF2      MOV R6,#-14 ; Moins de 14 erreurs admissibles pour la séquence double
   0177 B67D      JF0 L017D

      000C =         Tsize SET 6 * 2 ; Table de 6 lignes de 2 octets pour la séquence simple
    0179 B93C      MOV R1,#RAM030 + Tsize ; Sinon l'assembleur CYS8049 bogue sur #RAM030+(6×2)
      017B BEFA      MOV R6,#-6 ; Moins de 6 erreurs admissibles pour la séquence simple

                L017D:
      017D C9        DEC R1 ; Lignes 622 ignorées
   017E C9        DEC R1
      017F C9        DEC R1 ; Etats ligne 310 des trames impaires de même n°
   0180 FB        MOV A,R3
      0181 3286      JB1 L0186 ; Cas où 1 seul bit conservé, R3 = 00000001B (L04A3)
      0183 F1        MOV A,@R1 ; Lorsqu'appelé par traitement en début trame n° 1
      0184 248B      JMP L018B ; Pas de vote majoritaire pour un seul bit

                ;
                L0186:

      0186 F1        MOV A,@R1 ; Bits d'état de la ligne 310 concernée

      0187 5B        ANL A,R3 ; Elimination des bits de gauche inutiles
     0188 03B9      ADD A,#LOW L01B9 ; Table de vote majoritaire
      018A A3        MOVP A,@A ; Etat voté

                L018B:
      018B 67        RRC A ; Cy = état voté ou état entré lors dernière séquence
   018C F0        MOV A,@R0
      018D F7        RLC A ; Insertion de cet état par la droite
   018E A0        MOV @R0,A
   018F 97        CLR C
      0190 A7        CPL C ; Cy = 1
   0191 C9        DEC R1
      0192 F1        MOV A,@R1 ; Octet des bits de certification associés
   0193 37        CPL A
      0194 5B        ANL A,R3 ; Elimination des bits de gauche inutiles
      0195 C6A0      JZ L01A0 ; Cy = 1 si tous certifiés
      0197 BA05      MOV R2,#5 ; 5 bits à vérifier

                L0199:
      0199 67        RRC A ; Cy = bit de certification à vérifier
      019A E69D      JNC L019D ; Si le bit de certification est OK
      019C 1E        INC R6 ; Incrément du compteur d'erreurs

                L019D:
   019D EA99      DJNZ R2,L0199
      019F 97        CLR C ; Cy = 0 s'ils ne sont pas tous certifiés

                L01A0:
      01A0 C8        DEC R0 ; Octet cumulant la certification de chaque ligne 310
     01A1 F0        MOV A,@R0
      01A2 F7        RLC A ; Entrée du bit de certification par la droite
   01A3 A0        MOV @R0,A
      01A4 18        INC R0 ; Retour sur l'octet cumulant les états
   01A5 F9        MOV A,R1
     01A6 D330      XRL A,#RAM030 ; Début de table
      01A8 967D      JNZ L017D ; Boucle sur ligne 310 trame impaire suivante
      01AA FE        MOV A,R6 ; Compteur d'erreurs
   01AB 37        CPL A
      01AC F2B8      JB7 L01B8 ; Si nombre d'erreurs >= 6 ou >=14 selon la séquence
     01AE 2326      MOV A,#00100110B ; Motif pour séquence double
     01B0 B6B4      JF0 L01B4
     01B2 2306      MOV A,#00000110B ; Motif pour séquence simple

                L01B4:
      01B4 D0        XRL A,@R0 ; Comparaison
      01B5 A0        MOV @R0,A ; Le ou les bits différents sont à 1
      01B6 C8        DEC R0 ; Octet cumulant la certification de chaque ligne 310
      01B7 50        ANL A,@R0 ; Uniquement différences sur bits certifiés

                L01B8:
  01B8 83        RET

                ;
                 ; ******************************************************
                 ; Table de vote majoritaire
                 ; Etat voté = état identique d'au moins 3 bits sur les 5
                 ; qui adressent la table
                 ; ******************************************************

                L01B9:
   01B9 00 00 00  DB 0, 0, 0, 0

 01BC 00 
   01BD 00 00 00  DB 0, 0, 0, 1

 01C0 01 
   01C1 00 00 00  DB 0, 0, 0, 1

 01C4 01 
   01C5 00 01 01  DB 0, 1, 1, 1

 01C8 01 
   01C9 00 00 00  DB 0, 0, 0, 1

 01CC 01 
   01CD 00 01 01  DB 0, 1, 1, 1

 01D0 01 
   01D1 00 01 01  DB 0, 1, 1, 1

 01D4 01 
   01D5 01 01 01  DB 1, 1, 1, 1

 01D8 01 
                ;

                 ; *********************************************************************
                 ; Remplissage table états lignes 310/622 par ceux d'une séquence double
                 ; *********************************************************************

                ;
                L01D9:

   01D9 B930      MOV R1,#RAM030
      01DB BA06      MOV R2,#6 ; Motif 010.011 d'une séquence double décalé d'un pas
     01DD 2326      MOV A,#00100110B ; Décalé circulairement car état ligne 310 trame 12 sur bit 0

                L01DF:
     01DF B1FF      MOV @R1,#11111111B ; Certification de tous les bits d'état
      01E1 19        INC R1 ; Au tour de l'octet des bits d'état
      01E2 67        RRC A ; Bit du motif dans Cy
     01E3 B100      MOV @R1,#00000000B ; Tous les bits d'état à 0 si bit du motif à 0
   01E5 E6E9      JNC L01E9
     01E7 B1FF      MOV @R1,#11111111B ; Tous les bits d'état à 1 si bit du motif à 1

                L01E9:
      01E9 19        INC R1 ; Lignes 622 ignorées
   01EA 19        INC R1
      01EB 19        INC R1 ; Octet des bits de certification suivant
     01EC EADF      DJNZ R2,L01DF ; Initialisation des 6 paires d'octets
  01EE 83        RET

                ;
      0200           ORG 0200H ; Page 2

                ;
                 ; ********************************************

6



RT86.LST
The Cybernetic Micro Systems 8048 Family Assembler, Version 3.00   Page   7

                 ; Table des retards indexée par bits X, Z et Y
                 ; ********************************************

                ;
                     L0200: ; X Z Y

      0200 FF        DB R1804 ; 0 0 0 : retard 2 = 1804 ns
      0201 AF        DB R0000 ; 0 0 1 : retard 0 = vidéo directe
      0202 AF        DB R0000 ; 0 1 0 : retard 0 = vidéo directe
      0203 FF        DB R1804 ; 0 1 1 : retard 2 = 1804 ns
      0204 9F        DB R0902 ; 1 0 0 : retard 1 = 902 ns
      0205 AF        DB R0000 ; 1 0 1 : retard 0 = vidéo directe
      0206 FF        DB R1804 ; 1 1 0 : retard 2 = 1804 ns
      0207 9F        DB R0902 ; 1 1 1 : retard 1 = 902 ns

                ;
                 ; ********************************
                 ; Calcul du checksum sur la page 2
                 ; ********************************

                ;
                L0208:

      0208 FA        MOV A,R2 ; Adresse sur 8 bits dans la page
   0209 A3        MOVP A,@A
      020A F4ED      CALL L07ED ; Calcul de checksum sur 16 bits
   020C EA08      DJNZ R2,L0208
      020E 6400      JMP L0300 ; Poursuite du calcul sur la page 3

                 ;
                 ; ***************************************************
                 ; Déroulement des 286 lignes visibles d'une trame
                 ; Lecture état ligne 310/622 après défilement du bloc
                 ; En entrée :
                  ; R7 = numéro de trame
                  ; RAM04F.7
                  ; RAM04F.6
                 ; ***************************************************

                ;
                L0210:

     0210 B84F      MOV R0,#RAM04F ; Status de l'appareil
   0212 F0        MOV A,@R0
      0213 8820      ORL BUS,#00100000B ; DB5 à 1 : extinction LED verte
      0215 F219      JB7 L0219 ; Si synchronisé
   0217 443F      JMP L023F

                ;
                L0219:

      0219 D221      JB6 L0221 ; Si accès autorisé
   021B B92A      MOV R1,#RAM02A
      021D 11        INC @R1 ; Incrément temporisation de clignotement LED verte
   021E F1        MOV A,@R1 
      021F 9223      JB4 L0223 ; Inversion état LED toutes les 16 trames

                L0221:
     0221 98DF      ANL BUS,#11011111B ; Allumage LED verte

                L0223:
   0223 FF        MOV A,R7
   0224 5303      ANL A,#00000011B
      0226 AA        MOV R2,A ; de 1 à 3 itérations supplémentaires selon n° de trame
   0227 C62D      JZ L022D

                L0229:
      0229 54AE      CALL L02AE ; Itération du générateur pseudo aléatoire
   022B EA29      DJNZ R2,L0229

                ;
                 ; **************************************************
                 ; Obtention du bit Z
                 ; R7 = numéro de trame
                 ; RAM04F.4
                 ; Bit Z = R3.4 = 0 pour n° de trame [0..3] et [7..9] 
                 ; Bit Z = R3.4 = 1 pour n° de trame [4..6] et >= 10
                 ; **************************************************

                 ;
                L022D:

      022D BB00      MOV R3,#0 ; Bit Z à zéro par défaut
   022F FF        MOV A,R7
      0230 37        CPL A ; Complément à 1 du numéro de trame
   0231 0304      ADD A,#4
      0233 F63F      JC L023F ; Numéro de trame < 4
   0235 0303      ADD A,#3
      0237 F63D      JC L023D ; Numéro de trame < 7
   0239 0303      ADD A,#3
      023B F63F      JC L023F ; Numéro de trame < 10

                L023D:
     023D BB10      MOV R3,#00010000 ; Bit Z à 1

                L023F:
     023F BA1E      MOV R2,#286 MOD 256 ; 286 lignes retardables
   0241 B902      MOV R1,#(286 / 256) + 1
     0243 B84F      MOV R0,#RAM04F ; Status de l'appareil
   0245 F0        MOV A,@R0
      0246 9257      JB4 L0257 ; Si séquence brouillée

                ;
                 ; **********************************************************************
                 ; Déroulement d'un bloc de 286 lignes non retardées
                 ; En entrée :
                  ;  R1 = nombre de paquets de 256 lignes + 1 pour inclure le dernier
                  ; R2 = nombre de lignes du dernier paquet
                 ; **********************************************************************

                ;
                L0248:

   0248 42        MOV A,T
      0249 D317      XRL A,#23 ; Attente 23ème ligne (23 ou 335)

   024B 9648      JNZ L0248
                L024D:

      024D 464D      JNT1 L024D ; Attente top de synchro de la ligne suivante
                L024F:

      024F 564F      JT1 L024F ; Attente fin de ce top
   0251 EA4D      DJNZ R2,L024D
   0253 E94D      DJNZ R1,L024D
      0255 446E      JMP L026E ; Ligne 310 ou 622 atteinte

                ;
                 ; ****************************************************************************
                 ; Déroulement d'un bloc de 286 lignes retardées
                 ; A l'instar des articles parus dans Science et Vie 
                  ; N° 808 de Janvier 1985
                  ; N° 832 de Janvier 1987
                 ; Les bits qui déterminent les retards ont reçu la même dénomination X, Y et Z
                 ; En entrée :
                  ;  R1 = nombre de paquets de 256 lignes + 1 pour inclure le dernier
                  ; R2 = nombre de lignes du dernier paquet
                  ; R4/R5 : générateur pseudo aléatoire
                  ; RAM04F.0 = ?
                 ; ****************************************************************************

                ;
                L0257:

   0257 42        MOV A,T
      0258 D317      XRL A,#23 ; Attente 23ème ligne (23 ou 335)
   025A 9657      JNZ L0257

                L025C:
   025C FC        MOV A,R4
      025D F7        RLC A ; Bit X1 alias bit Y dans Cy
   025E FD        MOV A,R5
     025F 5320      ANL A,#00100000B ; Isolement du bit X11 alias le bit X
      0261 4B        ORL A,R3 ; Bit R3.5 = bit X
      0262 47        SWAP A ; A = [0.0.0.0.0.0.X.Z]
      0263 F7        RLC A ; A = [0.0.0.0.0.X.Z.Y]
      0264 A3        MOVP A,@A ; Obtention du retard approprié

                L0265:
      0265 4665      JNT1 L0265 ; Attente top de synchro de la ligne suivante
      0267 3A        OUTL P2,A ; Mise en place du retard obtenu
      0268 54AE      CALL L02AE ; Itération du générateur pseudo aléatoire
   026A EA5C      DJNZ R2,L025C
     026C E95C      DJNZ R1,L025C ; Ligne 309 ou 621 atteinte en fin de boucle

                L026E:
     026E 23FF      MOV A,#R1804 ; Retard 2 par défaut

                L0270:
      0270 4670      JNT1 L0270 ; Attente top de synchro ligne 310/622
   0272 3A        OUTL P2,A
     0273 987F      ANL BUS,#01111111B ; Temporisation minimale pour détection niveau du blanc
      0275 F0        MOV A,@R0 ; Status appareil
   0276 127A      JB0 L027A
   0278 447C      JMP L027C

                ;
                 ; **************************************************************
                 ; Détermination de l'état d'une ligne 310/622 par échantillonage
                 ; Les 4 échantillons issus du détecteur de niveau doivent être
                 ; identiques pour que l'état de la ligne soit certifié
                  ; En sortie :
                  ; R3.0 = état ligne, 0 ou 1
                  ; R3.7 = 1 si état certifié
                  ; sinon R3 = 0
                 ; **************************************************************

                ;
                L027A:

     027A 98EF      ANL BUS,#11101111B ; DB4 à zéro. Broche non connectée
                L027C:

   027C 2300      MOV A,#0
     027E BB80      MOV R3,#1 SHL 7 ; Ligne 310/622 certifiée à 0 par défaut
      0280 86A6      JNI L02A6 ; JMP L0284
   0282 17        INC A
  0283 00        NOP

                L0284:
      0284 00        NOP ; 2ème echantillonage
  0285 00        NOP
      0286 86A8      JNI L02A8 ; JMP L028A
   0288 17        INC A
  0289 00        NOP

                L028A:
      028A 00        NOP ; 3ème echantillonage
  028B 00        NOP
      028C 86AA      JNI L02AA ; JMP L0290
   028E 17        INC A
  028F 00        NOP

                L0290:
      0290 00        NOP ; 4ème echantillonage
  0291 00        NOP
      0292 86AC      JNI L02AC ; JMP L0296
   0294 17        INC A
  0295 00        NOP

                L0296:
  0296 00        NOP
     0297 8810      ORL BUS,#00010000B ; Retour DB4 à 1. Broche non connectée

                L0299:
      0299 4699      JNT1 L0299 ; Attente top de synchro ligne suivante
     029B 8880      ORL BUS,#10000000B ; Temporisation mormale pour détection niveau du blanc
      029D C6A5      JZ L02A5 ; Ligne certifiée à 0

    029F BB81      MOV R3,#(1 SHL 7) + 1 ; Ligne certifiée à 1
      02A1 52A5      JB2 L02A5 ; Les 4 échantillons doivent être identiques
      02A3 BB00      MOV R3,#0 ; Etat de la ligne non certifié, zéro par défaut

                L02A5:
  02A5 83        RET

                L02A6:
   02A6 4484      JMP L0284

                L02A8:
   02A8 448A      JMP L028A

                L02AA:
   02AA 4490      JMP L0290

                L02AC:
   02AC 4496      JMP L0296

                ;
                 ; **************************************************************************
                 ; Itération du générateur pseudo aléatoire dans R4 + R5
                 ; Polynôme  générateur P(X) = X11 + X9 + 1
                 ; Le générateur est implanté logiciellement sous la forme SSRG ou "Fibonacci"
                 ; 
                 ;        |<--------------R4------------->|<-------------R5-------------->|
                 ;    ,->-[ X1| X2| X3| X4| X5| X6| X7| X8| X9|X10|X11|...|...|...|...|...]
                 ;    |   bit Y                             |       |
                 ;    '-------<-------<-------<-------<----[×]--<---'   X11 = bit X
                 ;
                 ; ***************************************************************************

                ;
                L02AE:

   02AE FD        MOV A,R5
      02AF F7        RLC A ; Le bit X9 est dans Cy
   02B0 A7        CPL C
      02B1 D2B4      JB6 L02B4 ; OU exclusif avec le bit X11
   02B3 A7        CPL C

                L02B4:
   02B4 FC        MOV A,R4
      02B5 67        RRC A ; OU exclusif avec le bit X11
   02B6 AC        MOV R4,A
   02B7 FD        MOV A,R5
      02B8 67        RRC A ; Le bit X8 est propagé dans R5
   02B9 AD        MOV R5,A
  02BA 83        RET

                ;
                 ; *****************************************************************************
                 ; Système NON synchronisé :
                 ; Seulement traitement état lignes 310 en début trame paire
                 ; Cumul de ces états et de leur certification sur 16 bits à partir de RAM050
                 ;
                 ; |<------ RAM050+2 ----->|<------ RAM050+3 ----->|
                 ; [15|14|13|12|11|10|09|08|07|06|05|04|03|02|01|00]<---[bit état ligne 310]
                 ;   <----------------- Décalage <-----------------
                 ;
                 ; |<------- RAM050 ------>|<------ RAM050+1 ----->|
                 ; [15|14|13|12|11|10|09|08|07|06|05|04|03|02|01|00]<---[bit certification état]
                 ;   <----------------- Décalage <-----------------
                 ;
                 ; *****************************************************************************
                 ; Système synchronisé :
                 ; Traitement état lignes 310 en début trame paire
                 ; Traitement état lignes 622 en début trame impaire
                 ; Cumul de l'état de la ligne pour chaque trame sur 1 octet
                 ; Idem pour le bit de certification associé
                 ; Paire d'octets indexée par n° de trame (T × 2) dans table en RAM030
                 ;
                 ; |<- RAM030 +(T×2) + 1 ->|
                 ; [07|06|05|04|03|02|01|00]<---[bit état ligne]
                 ;   <----- Décalage <-----
                 ;
                 ; |<--- RAM030 +(T×2) --->|
                 ; [07|06|05|04|03|02|01|00]<---[bit certification état]
                 ;   <----- Décalage <-----
                 ; *****************************************************************************

                ;
                L02BB:

      02BB FF        MOV A,R7 ; Numéro de trame actuel, le n° zéro est transitoire
      02BC E7        RL A ; Sert d'index pour accéder ligne 310/622 de la trame précédente
     02BD 0331      ADD A,#RAM030 + 1 ; 
   02BF A9        MOV R1,A
     02C0 B84F      MOV R0,#RAM04F ; Status de l'appereil
   02C2 F0        MOV A,@R0
      02C3 F2CA      JB7 L02CA ; Si synchronisé sur séquence
   02C5 FF        MOV A,R7
      02C6 12E3      JB0 L02E3 ; Ligne 622 ignorée début trame impaire suivante
   02C8 B953      MOV R1,#RAM050 + 3

                L02CA:
      02CA FB        MOV A,R3 ; Etat/bit certification dernière ligne 310/622
      02CB 67        RRC A ; Etat dans Cy
      02CC F1        MOV A,@R1 ; RAM050 + 3 ou RAM030 + 1 + (n° trame × 2)
      02CD F7        RLC A ; Inclusion par la droite du dernier état dans @R1
   02CE A1        MOV @R1,A
   02CF F0        MOV A,@R0
      02D0 F2D6      JB7 L02D6 ; Pas de propagation sur 16 bits si synchronisé
      02D2 C9        DEC R1 ; 8 bits d'état de poids forts
      02D3 F1        MOV A,@R1 ; RAM050 + 2
      02D4 F7        RLC A ; Propagation dans l'octet de poids fort
   02D5 A1        MOV @R1,A

                L02D6:
      02D6 C9        DEC R1 ; Bits de certification de poids faible
      02D7 FB        MOV A,R3 ; Etat/bit certification dernière ligne 310/622
      02D8 F7        RLC A ; Bit de certification dans Cy
      02D9 F1        MOV A,@R1 ; RAM050 + 1 ou RAM030 + (n° trame × 2)
      02DA F7        RLC A ; Inclusion du dernier bit de certification dans @R1
   02DB A1        MOV @R1,A
   02DC F0        MOV A,@R0
      02DD F2E3      JB7 L02E3 ; Pas de propagation sur 16 bits si synchronisé
      02DF C9        DEC R1 ; 8 bits de certification de poids forts
      02E0 F1        MOV A,@R1 ; RAM050
      02E1 F7        RLC A ; Propagation dans l'octet de poids fort
   02E2 A1        MOV @R1,A

                L02E3:
  02E3 83        RET

                ;
                 ; **************************************
                 ; Scrutation du clavier sans transcodage
                 ; A = 0 si présence touche ENT
                 ; **************************************

                ;
                L02E4:

      02E4 B4BC      CALL L05BC ; Scrutation du clavier sans transcodage
      02E6 D30C      XRL A,#12 ; Code 12 = touche MEMO jaune
   02E8 96EC      JNZ L02EC
     02EA 8840      ORL BUS,#01000000B ; DB6 à 1 : extinction LED jaune 

                L02EC:
      02EC D304      XRL A,#4 ; A = 0 si touche ENT (code = 8)
  02EE 83        RET

                ;
                 ; ****************************************
                 ; Initialisation d'une zone de mémoire RAM
                 ; R0 : adresse de la zone
                 ; R2 : nombre d'octets à initialiser
                 ; ****************************************

                ;
                L02EF:

   02EF 27        CLR A
                L02F0:

   02F0 A0        MOV @R0,A
   02F1 18        INC R0
   02F2 EAF0      DJNZ R2,L02F0
  02F4 83        RET

                ;
      0300           ORG 0300H ; Page 3

                ;
                 ; ********************************
                 ; Calcul du checksum sur la page 3
                 ; ********************************

                ;
                L0300:

      0300 FA        MOV A,R2 ; Adresse sur 8 bits dans la page
   0301 A3        MOVP A,@A
      0302 F4ED      CALL L07ED ; Calcul de checksum sur 16 bits
   0304 EA00      DJNZ R2,L0300
      0306 8400      JMP L0400 ; Poursuite du calcul sur la page 4

                ;
                 ; ***********************************************************
                 ; Saisie d'un code périodique à 10 chiffres au clavier
                  ; RAM05F : Zone de mémorisation des 10 chiffres
                  ; RAM020 : code binaire correspondant sur 32 bits
                 ; A = 0 si CRC4 calculé = CRC4 inclus dans le code périodique
                 ; ***********************************************************

                ;
                L0308:

     0308 89C0      ORL P1,#11000000B ; Désactivation commutation lente et démodulation audio
     030A 8820      ORL BUS,#00100000B ; DB5 à 1 : extinction LED verte
      030C BC00      MOV R4,#0 ; Initialisation compteur de répétition pour la touche "ENT"
      030E B458      CALL L0558 ; Attente frappe d'une touche avec clignotement LED jaune
   0310 37        CPL A
      0311 9237      JB4 L0337 ; Si touche frappée autre que la touche ENT

                L0313:
      0313 B458      CALL L0558 ; Attente frappe d'une nouvelle touche
      0315 B237      JB5 L0337 ; Si la touche MEMO a été frappée
      0317 9213      JB4 L0313 ; Si la touche ENT a été frappée
     0319 B85F      MOV R0,#RAM05F ; Zone de mémorisation des chiffres frappés sur le clavier
      031B BC0A      MOV R4,#10 ; Code périodique à 10 chiffres
      031D FB        MOV A,R3 ; Récupération du chiffre correspondant à la touche
   031E 6426      JMP L0326

                ;
                L0320:

      0320 B458      CALL L0558 ; Attente frappe d'une nouvelle touche
      0322 9213      JB4 L0313 ; Si la touche ENT a été frappée
      0324 B235      JB5 L0335 ; Si la touche MEMO a été frappée

                L0326:
      0326 A0        MOV @R0,A ; Mémorisation du chiffre frappé
   0327 18        INC R0
     0328 EC20      DJNZ R4,L0320 ; Attente frappe des 10 chiffres
      032A B458      CALL L0558 ; Attente validation par touche MEMO
   032C 37        CPL A
      032D B235      JB5 L0335 ; Si touche frappée autre que la touche MEMO
      032F F45B      CALL L075B ; Conversion du code périodique en nombre binaire sur 32 bits
      0331 74AE      CALL L03AE ; Vérification de ce code périodique par le CRC4
      0333 C670      JZ L0370 ; Si code 10 chiffres OK

                L0335:
     0335 98BF      ANL BUS,#10111111B ; DB6 à 0 : allumage LED jaune

                L0337:
   0337 044F      JMP L004F

                ;
                 ; **********************************************************
                 ; Réception d'une commande extérieure sur la liaison série
                  ; Liaison half-duplex sur port P10/contact 1 de la DIN
                  ; Inititiée par un "break" (niveau 0 durée > 250 ms)
                 ; **********************************************************

                ;
                L0339:

      0339 09        IN A,P1 ; Liaison série état sur P10/contact 1 de la DIN
   033A 37        CPL A
      033B 1239      JB0 L0339 ; Attente disparition du break présent sur la liaison série
     033D B826      MOV R0,#RAM026 ; Zone réceptrice
      033F BC01      MOV R4,#1 ; 1 seul octet attendu
      0341 D4B2      CALL L06B2 ; Réception de cet octet en série sur P10
   0343 BB50      MOV R3,#LOW L0351 - 1

                L0345:
      0345 1B        INC R3 ; Numéro de commande suivant
      0346 FB        MOV A,R3 ; 6 numéros possibles
   0347 A3        MOVP A,@A
      0348 C6AC      JZ L03AC ; Zéro = marqueur de fin de table
   034A D0        XRL A,@R0
      034B 9645      JNZ L0345 ; Exploration de toute la table 
   034D FB        MOV A,R3
    034E 0306      ADD A,#L0357 - L0351 ; Indexation dans la table d'adresses
      0350 B3        JMPP @A ; Appel de la routine associée au numéro de commande

                ;
                L0351:

      0351 0A        DB 10 ; Réception des 10 octets d'un code périodique
      0352 06        DB 6 ; Réception de 6 octets
   0353 80        DB 80H
   0354 40        DB 40H
   0355 20        DB 20H
   0356 00        DB 0

                L0357:
   0357 5D        DB LOW L035D
   0358 88        DB LOW L0388
   0359 82        DB LOW L0382
   035A 9A        DB LOW L039A
   035B 9E        DB LOW L039E
   035C AC        DB LOW L03AC

                ;
                 ; **********************************************************
                 ; Numéro de commande = 10 : Attente de 10 octets/chiffres
                 ; Réception d'une clef périodique transmise en série sur P10
                 ; **********************************************************

                ;
                L035D:

      035D BC0A      MOV R4,#10 ; 10 octets attendus
     035F B85F      MOV R0,#RAM05F ; Zone destinatrice des octets reçus
      0361 D4B2      CALL L06B2 ; Réception de ces 10 octets transmis en série sur P10
      0363 F45B      CALL L075B ; Conversion du code périodique en nombre binaire sur 32 bits
      0365 74AE      CALL L03AE ; Vérification du code périodique reçu
   0367 B826      MOV R0,#RAM026
      0369 A0        MOV @R0,A ; Sauvegarde résultat comparaison des CRC4
      036A BC01      MOV R4,#1
      036C D4D1      CALL L06D1 ; Reémission de ce résultat de comparaison
      036E 9635      JNZ L0335 ; CRC4 inégaux

                L0370:
     0370 B923      MOV R1,#RAM020 + 3 ; Dernier octet de la clef 28 bits
   0372 F1        MOV A,@R1
      0373 47        SWAP A ; Les deux bits de poids faible de la clef sont à droite
     0374 5303      ANL A,#00000011B ; Ces 2 derniers bits donnent le numéro de cette clef
      0376 E7        RL A ; ×2, la sauvegarde d'une clef requiert 2 registres
      0377 0304      ADD A,#4 ; 1er registre de l'EEROM où écrire les 4 clefs possibles
      0379 AD        MOV R5,A ; 1er registre de la paire
     037A B820      MOV R0,#RAM020 ; Adresse de la clef
      037C BC02      MOV R4,#2 ; 2 mots : 4 octets de RAM020 à RAM020 + 3
      037E D41A      CALL L061A ; Sauvegarde de la clef dans l'EEROM, CRC4 y compris
   0380 041F      JMP L001F

                ;
                 ; ********************************************************************
                 ; Numéro de commande = 80H
                 ; Emission série sur P10 du contenu des registres 0, 1 et 2 de l'EEROM
                 ; ********************************************************************

                ;
                L0382:

      0382 D411      CALL L0611 ; Lecture de 3 mots à partir du registre 0 de l'EEROM
      0384 9494      CALL L0494 ; Encryptage du quartet de droite de chaque octet
      0386 6492      JMP L0392 ; Emission de ces 6 octets en série sur P10

                ;
                 ; **********************************************************************
                 ; Numéro de commande = 6
                 ; Programmation du numéro d'appareil dans registres 0, 1 et 2 de l'EEROM
                 ; **********************************************************************

                ;
                L0388:

      0388 BC06      MOV R4,#6 ; 6 octets attendus
     038A B86A      MOV R0,#RAM06A ; Zone destinatrice des octets reçus
      038C D4B2      CALL L06B2 ; Réception de ces 6 octets transmis en série sur P10
      038E D408      CALL L0608 ; Ecriture des 6 octets dans EEROM à partir registre 0

      0390 D411      CALL L0611 ; Lecture de ces mêmes octets qui viennent d'être écrits
                L0392:

      0392 BC06      MOV R4,#6 ; 6 octets à émettre
     0394 B86A      MOV R0,#RAM06A ; Zone où sont stockés les octets lus dans l'EEROM
      0396 D4D1      CALL L06D1 ; Emission de ces 6 octets en série sur P10
   0398 044F      JMP L004F

                ;
                 ; ************************
                 ; Numéro de commande = 40H
                 ; Test du clavier
                 ; ************************

                ;
                L039A:

      039A F408      CALL L0708 ; Frappe touches 1 à 9, 0, ENT et MEMO
   039C 044F      JMP L004F

                ;
                 ; ************************************************************
                 ; Numéro de commande = 20H
                 ; Emission série sur P10 des 4 clefs intermédiaires de 32 bits
                 ; ************************************************************

                ;
                L039E:

      039E BC08      MOV R4,#8 ; 8 mots/16 octets pour les 4 clefs
      03A0 BD04      MOV R5,#4 ; Registre de départ
     03A2 B870      MOV R0,#RAM070 ; Zone destinatrice des mots/octets extraits
      03A4 D44C      CALL L064C ; Lecture dans l'EEROM
      03A6 BC10      MOV R4,#16 ; 16 octets à émettre
     03A8 B870      MOV R0,#RAM070 ; Zone où ont été stockés les octets extraits de l'EEROM
      03AA D4D1      CALL L06D1 ; Emission de ces 16 octets en série sur P10

                ;
                 ; *****************************
                 ; Numéro de commande nul = zéro
                 ; *****************************

                 ;
                L03AC:

   03AC 044F      JMP L004F
                ;

                 ; ********************************************************************
                 ; Contrôle de validité du code périodique à 10 chiffres
                 ; A = 0 si CRC4 calculé = CRC4 inclus dans code périodique 10 chiffres
                 ; R3 = CRC4 calculé
                 ; ********************************************************************
                 ; Génération du CRC4 :
                  ; Réalisé à partir d'un LFSR de type MSRG ou "Galois"
                  ; Ce LFSR est implanté dans le quartet gauche de R3
                  ; Le polynome générateur est de la forme P(x) = x4 + x + 1
                 ; Le CRC4 est initialisé de fait avec les 4 bits du quartet de
                 ; gauche présent dans R3 avant la 1ère itération
                 ;
                  ;     |<-------- R3 --------->|<-------- R4 --------->|
                  ;     [b7|b6|b5|b4|b3|b2|b1|b0|b7|b6|b5|b4|b3|b2|b1|b0]
                  ; ,-<-[x4|x3|x2*x1*x0| <----------------------------- ]
                  ; |____________|__|
                 ;
                 ; ******************************************************************
                 ; Calcul du CRC4 sur les 28 bits de gauche du code intermédiaire
                 ;
                 ; Le générateur est initialisé de fait avec bits 31 à 28 du code
                  ; [<---- R3 = RAM020 ---->|<--- R4 = RAM020+1 --->|
                  ; [31|30|29|28|27|26|25|24|23|22|21|20|19|18|17|16]
                  ; [ Générateur]
                 ; Les bits 27 à 20 vont entrer dans le générateur lors du 1er cycle
                 ;
                 ; Après le 1er cycle de 8 itérations et rechargement R4 par RAM020+2
                  ; [<-------- R3 --------->|<--- R4 = RAM20+2 ---->|
                  ; [<- CRC 4 ->|19|18|17|16|15|14|13|12|11|10| 9| 8]
                 ; Les bits 19 à 12 vont entrer dans le générateur lors du 2ème cycle
                 ; 
                 ; Après le 2ème cycle de 8 itérations et rechargement R4 par RAM020+3
                  ; [<-------- R3 --------->|<--- R4 = RAM20+3 ---->|
                  ; [<- CRC 4 ->|11|10| 9| 8| 7| 6| 5| 4| 3| 2| 1| 0]
                 ; Les bits 11 à 4 vont entrer dans le générateur lors du 3ème cycle
                  ; 
                 ; Après le 3ème cycle de 8 itérations
                  ; [<-------- R3 --------->|
                  ; [<- CRC 4 ->| 3| 2| 1| 0]
                 ; Les 4 bits restants de 3 à 0 correspondent au CRC4 inclus dans le
                 ; code périodique à 10 chiffres
                 ; ******************************************************************

                ;
                L03AE:

     03AE B820      MOV R0,#RAM020 ; Adresse du code intermédiaire à 32 bits
   03B0 F0        MOV A,@R0
   03B1 AB        MOV R3,A
      03B2 BA03      MOV R2,#3 ; 3 cycles de 8 itérations

                L03B4:
   03B4 18        INC R0
   03B5 F0        MOV A,@R0
   03B6 AC        MOV R4,A
      03B7 B908      MOV R1,#8 ; 8 itérations par octet

                L03B9:
      03B9 97        CLR C ; Décalage à gauche sur 16 bits
   03BA FC        MOV A,R4
        03BB F7        RLC A ; Décalage octet de poids faible

   03BC AC        MOV R4,A
   03BD FB        MOV A,R3
      03BE F7        RLC A ; Décalage octet de poids fort
   03BF E6C3      JNC L03C3
     03C1 D330      XRL A,#00110000B ; Correspond au OU exclusif du générateur type "Galois"

                L03C3:
   03C3 AB        MOV R3,A
     03C4 E9B9      DJNZ R1,L03B9 ; Entrée du bit suivant dans le générateur
     03C6 EAB4      DJNZ R2,L03B4 ; Entrée des 8 bits suivants dans le générateur
      03C8 47        SWAP A ; CRC4 calculé sur les 28 bits de gauche dans quartet gauche de R3
   03C9 530F      ANL A,#00001111B
   03CB AB        MOV R3,A
      03CC D0        XRL A,@R0 ; CRC4 contenu dans code périodique à 10 chiffres sur quartet droit
     03CD 530F      ANL A,#00001111B ; Zéro si OK
  03CF 83        RET

                ;
                 ; ***************************************************************
                 ; Désencryptage d'une clef intermédiaire
                  ; La clef sur 26 bits cadrée à gauche est stockée en RAM020
                  ; Le n° d'appareil sur 6 chiffres est stocké en RAM06A
                 ; ***************************************************************

                ;
                L03D0:

     03D0 B870      MOV R0,#RAM06A + 6 ; RAM06A = n° d'appareil extrait de l'EEROM
                L03D2:

      03D2 C8        DEC R0 ; Du poids le plus faible vers le plus fort
   03D3 F0        MOV A,@R0
      03D4 E7        RL A ; Multiplication ×2 car quartet gauche à zéro
      03D5 17        INC A ; 1 rotation minimum
      03D6 AA        MOV R2,A ; Nombre de rotation à droite.
      03D7 B43E      CALL L053E ; Rotation par la droite des 26 bits de la clef stockée en RAM020
      03D9 BA03      MOV R2,#3 ; Transcodage des quartets de droite des 3 premiers octets
   03DB B920      MOV R1,#RAM020

                L03DD:
   03DD F1        MOV A,@R1
     03DE 530F      ANL A,#00001111B ; Seul le quartet de droite indexera la table
   03E0 03ED      ADD A,#LOW L03ED
   03E2 A3        MOVP A,@A
      03E3 31        XCHD A,@R1 ; Transcodage du quartet de droite
   03E4 19        INC R1
     03E5 EADD      DJNZ R2,L03DD ; Transcodage sur l'octet suivant
   03E7 F8        MOV A,R0
   03E8 D36A      XRL A,#RAM06A
   03EA 96D2      JNZ L03D2
  03EC 83        RET

                ;
                 ; Table de transcodage des quartets

                ;
                L03ED:

   03ED 05 08 02  DB 5, 8, 2, 0CH, 0, 7, 0DH, 6
 03F0 0C 00 07 0D 06 

   03F5 01 0B 09  DB 1, 0BH, 9, 0FH, 4, 0EH, 0AH, 3
 03F8 0F 04 0E 0A 03 
                ;

      0400           ORG 0400H ; Page 4
                ;

                 ; ********************************
                 ; Calcul du checksum sur la page 4
                 ; ********************************

                ;
                L0400:

      0400 FA        MOV A,R2 ; Adresse sur 8 bits dans la page
   0401 A3        MOVP A,@A
      0402 F4ED      CALL L07ED ; Calcul de checksum sur 16 bits
   0404 EA00      DJNZ R2,L0400
      0406 A400      JMP L0500 ; Poursuite du calcul sur la page 5

                ;
                 ; *****************************************************************
                 ; Traitement spécifique en début des trames n° 6 et n° 12
                  ; Autorisation d'accès selon niveau d'audience
                  ; Modification de la clef uniformisée selon niveau d'audience
                  ; Initialisation du générateur pseudo aléatoire
                 ; *****************************************************************
                 ; Clef uniformisée :
                 ; [<--- RAM020 -->|<-- RAM020+1 ->|<-- RAM020+2 ->|<-- RAM020+3 ->|
                 ; [3|3|2|2|2|2|2|2|2|2|2|2|×|×|×|×|1|1|1|1|1|1|0|0|0|0|×|×|1|1|1|1]
                 ; [1|0|9|8|7|6|5|4|3|2|1|0|0|0|0|0|5|4|3|2|1|0|9|8|7|6|0|0|9|8|7|6]
                 ; |<--- Clef 11 bits -->|^|        | | | | | | | | | | X | | | | |
                 ;          12ème bit ----'         1                     1 1 1 1 1
                 ; Niveau d'audience associé -----> 0 9 8 7 6 5 4 3 2 1 0 5 4 3 2 1
                  ; *****************************************************************

                ;
                L0408:

     0408 B92B      MOV R1,#RAM02B ; N° de clef à utiliser reçu dans la séquence
   040A F1        MOV A,@R1
     040B 0370      ADD A,#RAM070 ; Zone où sont stockées les 4 clefs uniformisées
   040D A9        MOV R1,A
     040E B820      MOV R0,#RAM020 ; Adresse de recopie de la clef
   0410 BA04      MOV R2,#4
      0412 94F5      CALL L04F5 ; Recopie de ces 4 octets
     0414 B84F      MOV R0,#RAM04F ; Status de l'appareil
     0416 B92C      MOV R1,#RAM02C ; Niveau d'audience reçu dans la séquence
   0418 F1        MOV A,@R1
      0419 AA        MOV R2,A ; R2 = niveau d'audience

      041A C678      JZ L0478 ; Si niveau d'audience nul
      041C D30F      XRL A,#15 ; Niveau audience 15 maximum pour séquence double
   041E B622      JF0 L0422
     0420 5307      ANL A,#00000111B ; Niveau audience 7 maximum pour séquence simple

                L0422:
      0422 C67D      JZ L047D ; Si niveau d'audience maximum
   0424 B923      MOV R1,#RAM020 + 3
      0426 F1        MOV A,@R1 ; Dernier octet des droits d'accès

                L0427:
      0427 67        RRC A ; Rotation circulaire des droits d'accès
   0428 C9        DEC R1
   0429 F1        MOV A,@R1
      042A 67        RRC A ; Rotation à droite de l'octet de poids supérieur
   042B A1        MOV @R1,A
   042C 19        INC R1
   042D F1        MOV A,@R1
      042E 67        RRC A ; Rotation à droite de l'octet de poids faible
   042F A1        MOV @R1,A
     0430 EA27      DJNZ R2,L0427 ; Autant de rotations que la valeur du niveau d'audience
      0432 B239      JB5 L0439 ; Accès à ce niveau d'audience autorisé

                ;
                 ; Accès interdit pour ce niveau d'audience

                ;
     0434 B080      MOV @R0,#10000000B ; Status avec accès au niveau d'audience interdit
     0436 8940      ORL P1,#01000000B ; Démodulateur audio désactivé
  0438 83        RET

                ;
                 ; Accès autorisé pour ce niveau d'audience

                ;
                L0439:

     0439 B92C      MOV R1,#RAM02C ; Niveau d'audience reçu dans la séquence
   043B F1        MOV A,@R1
      043C 9486      CALL L0486 ; Transcodage quartet droit octet de poids fort de la clef

                L043E:
     043E 99BF      ANL P1,#10111111B ; Mise en service démodulateur audio
     0440 B0D1      MOV @R0,#11010001B ; Nouveau status de l'appareil
   0442 B906      MOV R1,#6
   0444 BB6C      MOV R3,#LOW L046C
   0446 B848      MOV R0,#RAM048
   0448 BA00      MOV R2,#0

                L044A:
      044A FB        MOV A,R3 ; Lecture valeur à comparer
   044B A3        MOVP A,@A
      044C D0        XRL A,@R0 ; comparaison
      044D 18        INC R0 ; Ligne RAM suivante
      044E 1B        INC R3 ; Ligne valeur à ajouter
      044F 9655      JNZ L0455 ; Si comparaison NOK
   0451 FB        MOV A,R3
      0452 A3        MOVP A,@A ; Obtention valeur à ajouter
   0453 6A        ADD A,R2
      0454 AA        MOV R2,A ; Résultat addition dans R2

                L0455:
      0455 1B        INC R3 ; Ligne suivante de la table
   0456 E94A      DJNZ R1,L044A
      0458 FA        MOV A,R2 ; Résultat = décalage dans table de transcodage
      0459 C65D      JZ L045D ; Pas de transcodage si nul
      045B 9486      CALL L0486 ; Transcodage 

                L045D:
     045D B920      MOV R1,#RAM020 ; Adresse de la clef de 11/12 bits
      045F B804      MOV R0,#4 ; Adresse RAM 4 = R4, celle du générateur pseudo aléatoire
      0461 BA02      MOV R2,#2 ; 2 octets à transférer
      0463 94F5      CALL L04F5 ; Transfert des 2 octets de la clef dans R4/R5
      0465 C9        DEC R1 ; Octet de poids faible de la clef
   0466 F1        MOV A,@R1
      0467 926B      JB4 L046B ; Pas d'itération si 12ème bit de la clef est à 1
      0469 54AE      CALL L02AE ; Une itération immédiate sur le générateur

                L046B:
  046B 83        RET

                ;
                L046C:

   046C 03 01     DB 3, 1
   046E 04 02     DB 4, 2
   0470 02 03     DB 2, 3
   0472 01 01     DB 1, 1
   0474 05 02     DB 5, 2
   0476 04 03     DB 4, 3

                ;
                 ; ************************
                 ; Niveau d'audience = zéro
                 ; ************************

                ;
                L0478:

     0478 B0C1      MOV @R0,#11000001B ; Nouveau status de l'appareil
     047A 8940      ORL P1,#01000000B ; Démodulateur audio inactivé
  047C 83        RET

                ;
                ;

                 ; *************************
                 ; Niveau d'audience maximum
                 ; *************************

                ;
                L047D:

   047D B920      MOV R1,#RAM020
     047F B1A7      MOV @R1,#HIGH SESAME ; Clef universelle : 0A72CH

   0481 19        INC R1
   0482 B12C      MOV @R1,#LOW SESAME
   0484 843E      JMP L043E

                ;
                 ; ***********************************************************************
                 ; Transcodage du quartet droit dans l'octet gauche de la clef en fonction
                 ; du décalage dans A
                 ; ***********************************************************************

                ;
                L0486:

     0486 B920      MOV R1,#RAM020 ; Adresse octet de gauche de la clef
      0488 61        ADD A,@R1 ; Addition du décalage
     0489 530F      ANL A,#00001111B ; Le quartet de droite va servir d'index
     048B 03ED      ADD A,#LOW L03ED ; Table de transcodage
      048D E3        MOVP3 A,@A ; Obtention du quartet transcodé
   048E 9692      JNZ L0492
      0490 2307      MOV A,#7 ; Valeur zéro remplacée

                L0492:
      0492 31        XCHD A,@R1 ; Remplacement quartet droit dans octet gauche 
  0493 83        RET

                ;
                 ; *******************************************************
                 ; Encryptage d'octets à l'aide de la table de transcodage
                 ; *******************************************************

                ;
                L0494:

     0494 BBED      MOV R3,#LOW L03ED ; Table de transcodage
     0496 B86A      MOV R0,#RAM06A ; Le quartet de gauche doit être nul
      0498 BC06      MOV R4,#6 ; 6 octets à transcoder

                L049A:
   049A FB        MOV A,R3
      049B E3        MOVP3 A,@A ; Transcodage 
      049C D0        XRL A,@R0 ; Encryptage par OU exclusif
   049D A0        MOV @R0,A
      049E 1B        INC R3 ; Bug ? utilité ? car la table est trop courte
      049F 18        INC R0 ; Octet suivant
     04A0 EC9A      DJNZ R4,L049A ; Encryptage des 6 octets
  04A2 83        RET

                ;
                 ; *******************************************************
                 ; Traitement spécifique en début de trame n° 1. Utilité ?
                 ; *******************************************************

                ;
                L04A3:

   04A3 BB01      MOV R3,#1
      04A5 346B      CALL L016B ; Cumul états lignes 310 de la dernière séquence

                ;
                 ; Cumul en RAM026 de l'état des compteurs de la table en RAM048
                 ;
                 ; RAM026 : [7|6|5|4|3|2|1|0]
                 ;               | | | | | '---- RAM048 \
                 ;               | | | | '-------- +1    |     6 ×
                 ;               | | | '---------- +2     \ compteurs
                 ;               | | '------------ +3     /     à
                 ;               | '-------------- +4    |   rebours
                 ;               '---------------- +5   / 
                 ; Le bit de RAM026 est à 1 quand le compteur à rebours lié vient d'atteindre zéro

                
                ;

     04A7 B926      MOV R1,#RAM026 ; Cumul de l'état des 6 compteurs
   04A9 B100      MOV @R1,#0
     04AB B848      MOV R0,#RAM048 ; Début de la table des compteurs à rebours
   04AD BA06      MOV R2,#6

                L04AF:
      04AF 97        CLR C ; Cy = 0 par défaut
      04B0 F0        MOV A,@R0 ; Compteur à zéro
      04B1 C6B8      JZ L04B8 ; Bit associé à zéro si compteur à zéro
      04B3 07        DEC A ; Décrémentation du compteur à rebours associé
   04B4 A0        MOV @R0,A
      04B5 96B8      JNZ L04B8 ; Bit associé à zéro si compteur était différent de 1
      04B7 A7        CPL C ; Bit associé à 1 si compteur maintenant à 1

                L04B8:
   04B8 F1        MOV A,@R1
      04B9 67        RRC A ; Le bit associé entre par la gauche
   04BA A1        MOV @R1,A
      04BB 18        INC R0
     04BC EAAF      DJNZ R2,L04AF ; Table de 6 compteurs à contrôler  
   04BE 77        RR A
      04BF 77        RR A ; Cadrage à droite des 6 bits du cumul
      04C0 A1        MOV @R1,A ; Bit associé à 1 pour le compteur qui vient d'atteindre zéro

                ;
                 ; A chaque compteur est associé 1 bit demande de réinitialisation dans RAM026
                 ; A chaque ligne 310 de la séquence est affecté un compteur en propre

                ;
                ; .-<-----<-----<-----<-----<-----<-----<-----<-----<-----<-----<-----<-----<--.
                ; |                                                                            |
                ; |      6|    3|    6|    3|    6|    3|    6|    3|    6|    3|    6|    3|  |
                ; '->T12 2|T11 1|T10 2|T9  1|T8  2|T7  1|T6  2|T5  1|T4  2|T3  1|T2  2|T1  1|T0'
                ;        2|    0|    2|    0|    2|    0|    2|    0|    2|    0|    2|    0|
                ;       '---' '---' '---' '---' '---' '---' '---' '---' '---' '---' '---' '---'
                ; RAM050: |   bit 5   |   bit 4   |   bit 3   |   bit 2   |   bit 1   |   bit 0
                ; Compteur|  RAM04D   |  RAM04C   |  RAM04B   |  RAM04A   |  RAM049   |  RAM048
                ;         |           |           |           |           |           |    
                ; RAM052:bit 5      bit 4       bit 3       bit 2       bit 1       bit 0

                ;
                ;

                 ; Cumul des bits non certifiés,
                 ; Le compteur associé ne sera rechargé que s'il vient d'atteindre zéro

                 ;
     04C1 B850      MOV R0,#RAM050 ; Adresse certificats état lignes 310 dernière séquence
      04C3 F0        MOV A,@R0 ; Bits de certification
      04C4 37        CPL A ; Ne seront rechargés que les compteurs des lignes 310 non certifiées
        04C5 51        ANL A,@R1 ; Et uniquement ceux venant juste d'atteindre zéro  
   04C6 AE        MOV R6,A 

                ;
                 ; Cumul des bits certifiés à 1,
                 ; rechargent systématiquement le compteur à rebours associé

                ;
      04C7 F0        MOV A,@R0 ; Bits de certification
      04C8 18        INC R0 ; Bits d'état
      04C9 50        ANL A,@R0 ; Ne seront à 1 que les bits certifiés à 1
      04CA 4E        ORL A,R6 ; Compteur aussi rechargé si la ligne 310 associée est certifiée à 1
   04CB AE        MOV R6,A

                ;
                 ; Cumul des bits certifié à zéro,
                 ; resettent systématiquement le compteur à rebours associé

                 ;
      04CC F0        MOV A,@R0 ; R0 = adresse états lignes 310 dernière séquence
      04CD 37        CPL A ; Inversion des états
      04CE C8        DEC R0 ; R0 = adresse certificats de ces lignes
      04CF 50        ANL A,@R0 ; Ne seront resetté que les compteurs des lignes 310 certifiées à 0
      04D0 51        ANL A,@R1 ; Et uniquement ceux venant juste d'atteindre zéro
   04D1 A1        MOV @R1,A

                ;
                 ; 1er opération : rechargement du compteur si le bit associé est à 1 dans R6
                 ;
                 ; R6 : [7|6|5|4|3|2|1|0]         RAM048
                 ;           | | | | | '-- 04H ->--- 0 --<- 0 ------------.
                 ;           | | | | '---- 08H ->-- +1 --<- 0 ----------. |
                 ;           | | | '------ 10H ->-- +2 --<- 0 --------. | |
                 ;           | | '-------- 20H ->-- +3 --<- 0 ------. | | |
                 ;           | '---------- 40H ->-- +4 --<- 0 ----. | | | | 
                 ;           '------------ 7FH ->-- +5 --<- 0 --. | | | | |
                 ;                                              | | | | | |
                 ;                                RAM026 : [7|6|5|4|3|2|1|0]
                 ;
                 ; 2ème opération : reset du compteur si le bit associé est à 1 dans RAM026

                ;
     04D2 B848      MOV R0,#RAM048 ; Début de la table des compteurs à rebours
      04D4 BA06      MOV R2,#6 ; 6 compteurs
   04D6 FE        MOV A,R6

                L04D7:
   04D7 FE        MOV A,R6
      04D8 67        RRC A ; Lecture des 6 bits de droite de R6 
   04D9 AE        MOV R6,A
      04DA E6E4      JNC L04E4 ; Pas d'initialisation du compteur si bit à zéro
   04DC FA        MOV A,R2
    04DD 03EE      ADD A,#LOW L04EF - 1 ; Indexation sur la valeur à recharger
   04DF A3        MOVP A,@A
      04E0 B6E3      JF0 L04E3 ; Pas d'opération si séquence double
      04E2 E7        RL A ; Multiplication de la valeur par 2 si séquence double

                L04E3:
      04E3 A0        MOV @R0,A ; Rechargement compteur si bit associé à 1 dans R6

                L04E4:
   04E4 F1        MOV A,@R1
      04E5 67        RRC A ; Lecture des 6 bits de droite de RAM026
   04E6 A1        MOV @R1,A
      04E7 E6EB      JNC L04EB ; Pas d'initialisation du compteur si bit à zéro
      04E9 B000      MOV @R0,#0 ; Reset du compteur si bit associé à 1 dans RAM026

                L04EB:
      04EB 18        INC R0
     04EC EAD7      DJNZ R2,L04D7 ; Intervention sur les 6 compteurs
  04EE 83        RET

                ;
                L04EF:

      04EF 7F        DB 7FH ; (2^7) -1
      04F0 40        DB 40H ; 2^6
      04F1 20        DB 20H ; 2^5
      04F2 10        DB 10H ; 2^4
      04F3 08        DB 8 ; 2^3
      04F4 04        DB 4 ; 2^2

                ;
                 ; ******************************
                 ; Recopie d'octets
                  ; Adresse origine     : R1
                  ; Adresse destination : R0
                  ; Nombre              : R2
                 ; ******************************

                ;
                L04F5:

   04F5 F1        MOV A,@R1
   04F6 A0        MOV @R0,A
   04F7 19        INC R1
   04F8 18        INC R0
   04F9 EAF5      DJNZ R2,L04F5
  04FB 83        RET

                ;
      0500           ORG 0500H ; Page 5

                ;
                 ; ********************************
                 ; Calcul du checksum sur la page 5
                 ; ********************************

                ;
                L0500:

      0500 FA        MOV A,R2 ; Adresse sur 8 bits dans la page
   0501 A3        MOVP A,@A
      0502 F4ED      CALL L07ED ; Calcul de checksum sur 16 bits
   0504 EA00      DJNZ R2,L0500
      0506 C400      JMP L0600 ; Poursuite du calcul sur la page 6

                ;
                 ; *****************************************************
                 ; Traitement spécifique en début de trame n° 3
                 ; Décodage du niveau d'audience à partir des lignes 622
                 ; *****************************************************

                ;
                L0508:

     0508 B853      MOV R0,#RAM052 + 1 ; Cumul états certifiés des lignes 622
   050A B000      MOV @R0,#0

      0018 =         Tsize SET 12 * 2 ; Table de 12 lignes de 2 octets, si séquence double
    050C B948      MOV R1,#RAM030 + Tsize ; Sinon l'assembleur CYS8049 bogue sur #RAM030+(12×2)
   050E B612      JF0 L0512

      000C =         Tsize SET 6 * 2 ; Table de 6 lignes de 2 octets, si séquence double
    0510 B93C      MOV R1,#RAM030 + Tsize ; Sinon l'assembleur CYS8049 bogue sur RAM030+(6×2)

                L0512:
      0512 C9        DEC R1 ; Octet états ligne 622 des dernières trames de même n°
   0513 97        CLR C
      0514 F1        MOV A,@R1 ; Etat de ces lignes 622
   0515 530F      ANL A,#00001111B
      0517 C61E      JZ L051E ; Cy = 0 si toutes les 4 à zéro
   0519 D30F      XRL A,#00001111B
      051B 963D      JNZ L053D ; Sortie si états non uniformes
      051D A7        CPL C ; Cy = 1 si toules 4 à 1

                L051E:
   051E F0        MOV A,@R0
      051F F7        RLC A ; Insertion de cet état par la droite
   0520 A0        MOV @R0,A
   0521 C9        DEC R1
      0522 F1        MOV A,@R1 ; Octet des bits de certification associés
   0523 37        CPL A
   0524 530F      ANL A,#00001111B
      0526 963D      JNZ L053D ; Sortie s'ils ne sont pas tous les 4 certifiés
   0528 C9        DEC R1
      0529 C9        DEC R1 ; Lignes 310 ignorées
   052A F9        MOV A,R1
     052B D330      XRL A,#RAM030 ; Début de table
      052D 9612      JNZ L0512 ; Boucle sur ligne 622 trame paire suivante
   052F F0        MOV A,@R0
     0530 5330      ANL A,#00110000B ; N° de clef sur 2 bits de droite du quartet gauche
      0532 77        RR A ; N° de clef à zéro pour séquence simple
      0533 77        RR A ; Division par 4 : index = (n° de clef × 4)
     0534 B92B      MOV R1,#RAM02B ; Sauvegarde de l'index en RAM02B
      0536 A1        MOV @R1,A ; 4 octets par clef
      0537 F0        MOV A,@R0 ; Cumul états certifiés des lignes 622
   0538 530F      ANL A,#00001111B
     053A B92C      MOV R1,#RAM02C ; Niveau d'audience en RAM02C
   053C A1        MOV @R1,A

                L053D:
  053D 83        RET

                ;
                 ; ************************************************************
                 ; Rotation par la droite des 26 bits stockés à partir de RAM20
                  ; Bits cadrés à gauche de 31 à 6
                  ; R2 = nombre de rotations à effectuer
                 ; ************************************************************

                ;
                L053E:

      053E FA        MOV A,R2 ; Nombre de rotations à droite
      053F C657      JZ L0557 ; Sortie immédiate si aucune rotation

                L0541:
     0541 B923      MOV R1,#RAM020 + 3 ; Bits de 0 à 7
      0543 F1        MOV A,@R1 ; Manipulation pour chasser le bit 6 dans Cy
   0544 E7        RL A
   0545 F7        RLC A
     0546 B920      MOV R1,#RAM020 ; Adresse des bits de 24 à 31
      0548 BB03      MOV R3,#3 ; Décalage à droite sur 3 × 8 bits

                L054A:
      054A F1        MOV A,@R1 ; Décalage à droite des bits 31 à 8
      054B 67        RRC A ; Le bit 6 entre à gauche et remplace le bit 31 
      054C A1        MOV @R1,A ; Le bit 8 est éjecté dans Cy
   054D 19        INC R1
   054E EB4A      DJNZ R3,L054A
      0550 F1        MOV A,@R1 ; Octet de poids faible
      0551 67        RRC A ; Le bit 8 entre à gauche et remplace le bit 7
   0552 53C0      ANL A,#11000000B
      0554 A1        MOV @R1,A ; On ne conserve que les deux bits de gauche
     0555 EA41      DJNZ R2,L0541 ; Rotation à droite suivante

                L0557:
  0557 83        RET

                ;
                 ; **********************************************************
                 ; Attente frappe d'une touche avec clignotement LED jaune
                 ; Retour avec F1 = 1

                  ; A = R3 = RAM029 = valeur/chiffre associé à la touche
                  ; F1 = 1 = touche acceptée
                 ; **********************************************************

                ;
                L0558:

      0558 BE00      MOV R6,#0 ; Séquentiel scrutation du clavier au repos
      055A BF1E      MOV R7,#30 ; Temporisation pour un cycle de clignotement LED jaune

                L055C:
      055C BA1B      MOV R2,#27 ; Temporisation

                L055E:
   055E E95E      DJNZ R1,L055E
   0560 EA5E      DJNZ R2,L055E
      0562 CF        DEC R7 ; Temporisation de clignotement
   0563 FF        MOV A,R7
      0564 966A      JNZ L056A ; Rechargement R7 pour un nouveau cycle
   0566 BF1E      MOV R7,#30
     0568 98BF      ANL BUS,#10111111B ; DB6 à 0, allumage LED jaune

                L056A:
      056A D30F      XRL A,#15 ; Inversion état LED jaune à moitié de cycle 
   056C 9670      JNZ L0570
     056E 8840      ORL BUS,#01000000B ; DB6 à 1, extinction LED jaune

                L0570:
      0570 B479      CALL L0579 ; Scrutation du clavier
      0572 7676      JF1 L0576 ; Une touche vient d'être frappée sur le clavier
      0574 A45C      JMP L055C ; Rebouclage

                ;
                L0576:

     0576 8840      ORL BUS,#01000000B ; DB6 à 1, extinction LED Jaune
  0578 83        RET

                ;
                 ; ************************************************************************
                 ; Séquentiel de scrutation du clavier dans R6
                  ; 0 = repos
                  ; 1 = touche présente
                  ; bit 7 à 1 = touche disparue (80H à 83H)
                 ; La touche sera validée une fois sa disparition certifiée 3 fois de suite
                  ; A et/ou R3 et/ou (RAM029) = chiffre associé à la touche
                  ; F1 = 1 si touche acceptée
                 ; ************************************************************************

                ;
                L0579:

      0579 B4BC      CALL L05BC ; Lecture du clavier, n° de touche dans R3 et dans A
      057B A5        CLR F1 ; Flag absence touche transcodée
   057C B929      MOV R1,#RAM029
      057E F2AD      JB7 L05AD ; Sortie si présence simultanée de plusieurs touches
      0580 C696      JZ L0596 ; Test si absence ou disparition touche
      0582 FE        MOV A,R6 ; Position du séquentiel
      0583 F2AF      JB7 L05AF ; Inchangée si réapparition après une disparition fugitive
      0585 C68B      JZ L058B ; Apparition d'une nouvelle touche quand séquentiel au repos
      0587 F1        MOV A,@R1 ; Comparaison avec n° de la touche précédente
   0588 DB        XRL A,R3
      0589 96AD      JNZ L05AD ; Parasitage si différent, séquentiel forcé à zéro

                L058B:
      058B FB        MOV A,R3 ; Apparition d'une nouvelle touche
      058C A1        MOV @R1,A ; Sauvegarde du n° de cette touche en RAM029
      058D 1E        INC R6 ; Position de séquentiel suivante
   058E FE        MOV A,R6
      058F D302      XRL A,#2 ; Test si position 2 atteinte
   0591 9695      JNZ L0595
      0593 BE01      MOV R6,#1 ; Retour en position 1 dans ce cas

                L0595:
  0595 83        RET

                ;
                L0596:

   0596 FE        MOV A,R6
      0597 F2A0      JB7 L05A0 ; Test si la touche vient juste de disparaître
      0599 D301      XRL A,#1 ; Test si séquentiel en position 1
      059B 96AD      JNZ L05AD ; Forçage séquentiel au repos même s'il l'est déjà
     059D BE80      MOV R6,#1 SHL 7 ; Bit 7 à 1 si la touche vient juste de disparaîtr
  059F 83        RET

                ;
                L05A0:

      05A0 1E        INC R6 ; Relecture du séquentiel
   05A1 FE        MOV A,R6
    05A2 D383      XRL A,#(1 SHL 7) + 3 ; La disparition doit être constatée 3 fois de suite
   05A4 96AF      JNZ L05AF
      05A6 B5        CPL F1 ; Le n° de touche a été transcodé en chiffre
    05A7 23AE      MOV A,#LOW L05B0 - 2 ; Transcodage du n° de touche en valeur numérique
   05A9 61        ADD A,@R1
      05AA A3        MOVP A,@A ; Transcodage du n° de touche en valeur numérique
      05AB A1        MOV @R1,A ; Chiffre correspondant à la dernière touche dans RAM029
      05AC AB        MOV R3,A ; Chiffre/valeur de la touche dans R3 et dans A

                L05AD:
      05AD BE00      MOV R6,#0 ; Séquentiel au repos

                L05AF:
  05AF 83        RET

                ;
                 ; *****************************************************
                 ; Table de transcodage n° de touche / valeur numérique
                 ; La table a pu être réduite de 16 à 12 lignes car :
                  ; N° 0 = abscence de touche
                  ; N° 1 = offset de départ pour scruter le clavier
                  ; Les n° 14 et 15 ne seront jamais indexés
                 ; *****************************************************

                ;
                L05B0:

      05B0 02        DB 2 ; N°  2 : chiffre 2
      05B1 05        DB 5 ; N°  3 : chiffre 5
      05B2 00        DB 0 ; N°  4 : chiffre 0
      05B3 08        DB 8 ; N°  5 : chiffre 8
      05B4 01        DB 1 ; N°  6 : chiffre 1
      05B5 04        DB 4 ; N°  7 : chiffre 4
      05B6 10        DB ENT ; N°  8 : touche "ENT"
      05B7 07        DB 7 ; N°  9 : chiffre 7
      05B8 03        DB 3 ; N° 10 : chiffre 3
      05B9 06        DB 6 ; N° 11 : chiffre 6
      05BA 20        DB MEMO ; N° 12 : touche "MEMO" jaune
      05BB 09        DB 9 ; N° 13 : chiffre 9

                ;
                 ; ******************************************
                 ; Disposition du clavier
                    ; .   .   .
                   ; P20--<--2---1---3- rangée n°1
                   ; |   |   |
                   ; P21--<--5---4---6- rangée n°2
                   ; |   |   |
                   ; P22--<--0--ENT-MEMO rangée n°3
                   ; |   |   |
                   ; P23--<--8---7---9- rangée n°4
                   ; |   |   |
                  ; P13-->--'   |   | - - - 1ère colonne 
                   ;     |   |
                  ; P14-->------'   | - - - 2ème colonne
                   ;         |
                  ; P15-->----------' - - - 3ème colonne
                 ;
                 ; ******************************************

                ;
                 ; ***********************************************************
                 ; Scrutation du clavier sans transcodage du n° de touche
                  ; A , R3 = n° de touche, 
                  ; bit 7 à 1 si présence simultanée de plusieurs touches
                 ; ***********************************************************

                ;
                L05BC:

     05BC BAF7      MOV R2,#11110111B ; Masque tournant, colonne P13 à zéro
      05BE B901      MOV R1,#1 ; Offset pour la 1ère ligne
   05C0 BB00      MOV R3,#0
      05C2 A5        CLR F1 ; Indicateur absence touche

                L05C3:
     05C3 8938      ORL P1,#00111000B ; Forçage à 1 des 3 colonnes (P13 à P15)
   05C5 09        IN A,P1
      05C6 5A        ANL A,R2 ; Forçage d'un zéro sur la colonne concernée
     05C7 4307      ORL A,#00000111B ; Forçage à 1 ports P10 à P12
   05C9 39        OUTL P1,A
      05CA 0A        IN A,P2 ; Lecture des rangées de touches
     05CB 8938      ORL P1,#00111000B ; Forçage à 1 des 3 colonnes (P13 à P15)
   05CD 530F      ANL A,#00001111B
     05CF 03EA      ADD A,#LOW L05EA ; Transcodage présence touche(s)
      05D1 A3        MOVP A,@A ; N° de rangée de la touche présente
      05D2 F2E6      JB7 L05E6 ; Sortie si présence simultanée de plusieurs touches
      05D4 C6DB      JZ L05DB ; Colonne suivante si absence touche
      05D6 76E6      JF1 L05E6 ; Sortie si présence d'une touche sur une colonne précédente
      05D8 B5        CPL F1 ; Indicateur à 1 si présence d'une touche
      05D9 69        ADD A,R1 ; Ajout de l'offset au n° de rangée
   05DA AB        MOV R3,A

                L05DB:
   05DB F9        MOV A,R1
      05DC 0304      ADD A,#4 ; Itération de l'offset pour la colonne suivante
   05DE A9        MOV R1,A
   05DF FA        MOV A,R2
      05E0 E7        RL A ; Rotation du masque sur colonne suivante
   05E1 AA        MOV R2,A
      05E2 D2C3      JB6 L05C3 ; Sortie si exploration des 3 colonnes terminée

                L05E4:
      05E4 FB        MOV A,R3 ; Retour avec n° de touche dans R3 et dans A
  05E5 83        RET

                L05E6:
     05E6 BB80      MOV R3,#1 SHL 7 ; Présence simultanée de plusieurs touches
   05E8 A4E4      JMP L05E4

                ;
                 ; ***********************************************************
                 ; Table de présence touche(s)
                  ; Bit 7 à 1 si plusieurs touches appuyées simultanément
                 ; ***********************************************************

                ;
                L05EA:

   05EA 80        DB 1 SHL 7
   05EB 80        DB 1 SHL 7
   05EC 80        DB 1 SHL 7
   05ED 80        DB 1 SHL 7
   05EE 80        DB 1 SHL 7
   05EF 80        DB 1 SHL 7
   05F0 80        DB 1 SHL 7
      05F1 04        DB 4 ; Rangée P23 seule à zéro 
   05F2 80        DB 1 SHL 7
   05F3 80        DB 1 SHL 7
   05F4 80        DB 1 SHL 7

      05F5 03        DB 3 ; Rangée P22 seule à zéro
   05F6 80        DB 1 SHL 7
      05F7 02        DB 2 ; Rangée P21 seule à zéro
      05F8 01        DB 1 ; Rangée P20 seule à zéro
      05F9 00        DB 0 ; Aucune touche

                ;
      0600           ORG 0600H ; Page 6

                ;
                 ; ********************************
                 ; Calcul du checksum sur la page 6
                 ; ********************************

                 ;
                L0600:

      0600 FA        MOV A,R2 ; Adresse sur 8 bits dans la page
   0601 A3        MOVP A,@A
      0602 F4ED      CALL L07ED ; Calcul de checksum sur 16 bits
   0604 EA00      DJNZ R2,L0600
      0606 E400      JMP L0700 ; Poursuite du calcul sur la page 7

                ;
                 ; ************************************
                 ; Jeu d'instruction de l'EEROMN MC9306
                 ; ************************************

                ;
      0080 =         READ EQU 80H ; Read Register

      0040 =         WRITE EQU 40H ; Write Register
      00C0 =         ERASE EQU 0C0H ; Erase Register

      0030 =         EWEN EQU 30H ; Erase/Write Enable
      0000 =         EWDS EQU 0 ; Erase/Write Disable

                ;
                 ; *********************************************************
                 ; Ecriture dans les 3 premiers registres de l'EEROM
                 ; Valeurs à écrire stockées à partir de RAM06A sur 6 octets
                 ; *********************************************************

                ;
                L0608:

     0608 B86A      MOV R0,#RAM06A ; Adresse stockage 6 octets/chiffres à écrire
      060A BC03      MOV R4,#3 ; 3 mots/6 octets à écrire
      060C BD00      MOV R5,#0 ; Ecriture à partir du registre 0 de l'EEROM
      060E D41A      CALL L061A ; Ecriture dans les registres de l'EEROM
  0610 83        RET

                ;
                 ; **********************************************************
                 ; Lecture des 3 premiers registres de l'EEROM
                 ; Valeurs extraites stockées à partir de RAM06A sur 6 octets
                 ; **********************************************************

                ;
                L0611:

      0611 BD00      MOV R5,#0 ; Lecture à partir du registre 0 de l'EEROM
     0613 B86A      MOV R0,#RAM06A ; Adresse stockage 6 octets/chiffres lus
      0615 BC03      MOV R4,#3 ; 3 mots/6 octets à lire
      0617 D44C      CALL L064C ; Lecture dans les registres de l'EEROM
  0619 83        RET

                ;
                 ; ****************************************************
                 ; Ecriture d'une suite de mots de 16 bits dans l'EEROM
                  ; R0 = adresse mémoire des mots à écrire
                  ; R4 = nombre de mots à écrire
                  ; R5 = numéro du premier registre à écrire
                 ; ****************************************************
                 ; Retour avec
                  ; R0 : adresse du dernier mot émis
                  ; R4 = 1
                  ; R5 = numéro du dernier registre écrit
                 ; ****************************************************

                ;
                L061A:

   061A BB30      MOV R3,#EWEN
      061C D466      CALL L0666 ; Emission de cet ordre
     061E 98F8      ANL BUS,#11111000B ; DI, CK et CS forcés à zéro

                L0620:
      0620 FD        MOV A,R5 ; Numéro du registre de l'EEROM
   0621 43C0      ORL A,#ERASE
      0623 AB        MOV R3,A ; Registre à effacer
   0624 D466      CALL L0666
     0626 98F8      ANL BUS,#11111000B ; DI, CK et CS forcés à zéro
      0628 D4AB      CALL L06AB ; Temporisation d'écriture ~20 ms
      062A FD        MOV A,R5 ; Récupération du numéro de registre
   062B 4340      ORL A,#WRITE
      062D AB        MOV R3,A ; Registre à écrire
   062E D466      CALL L0666
      0630 F0        MOV A,@R0 ; Récupération octet de poids fort
   0631 18        INC R0
      0632 D478      CALL L0678 ; Ecriture dans bits D15 à D8 de l'EEROM 
      0634 F0        MOV A,@R0 ; Récupération octet de poids faible
      0635 D478      CALL L0678 ; Ecriture dans bits D7 à D0 de l'EEROM
     0637 98FD      ANL BUS,#11111101B ; SK à 0
     0639 98FB      ANL BUS,#11111011B ; CS à 0
      063B D4AB      CALL L06AB ; Temporisation d'écriture ~20 ms
      063D 18        INC R0 ; Adresse octet/mot suivant
      063E 1D        INC R5 ; Registre suivant
     063F EC20      DJNZ R4,L0620 ; Ecriture de tous les mots de 16 bits
      0641 1C        INC R4 ; R4 = 1
   0642 C8        DEC R0
      0643 C8        DEC R0 ; R0 : adresse du dernier mot émis

      0644 CD        DEC R5 ; R5 = numéro dernier registre écrit
   0645 BB00      MOV R3,#EWDS
      0647 D466      CALL L0666 ; Emission de ce dernier ordre
     0649 98F8      ANL BUS,#11111000B ; DI, CK et CS forcés à zéro
  064B 83        RET

                ;
                 ; ******************************************************
                 ; Lecture d'une suite de mots de 16 bits dans l'EEROM
                  ; R0 = adresse où stocker les mots qui ont été lus
                  ; R4 = nombre de mots à lire
                  ; R5 = numéro du premier registre à lire
                 ; ******************************************************
                 ; Retour avec :
                  ; R0 : adresse du dernier mot reçu
                  ; R4 = 1
                  ; R5 = numéro du dernier registre lu
                 ; ******************************************************

                ;
                L064C:

      064C FD        MOV A,R5 ; Numéro du registre de l'EEROM
   064D 4380      ORL A,#READ
      064F AB        MOV R3,A ; Registre à lire
   0650 D466      CALL L0666
      0652 D49C      CALL L069C ; Lecture du "dummy" bit
      0654 D480      CALL L0680 ; Lecture 1er octet
      0656 A0        MOV @R0,A ; Bits D15 à D8 du registre
   0657 18        INC R0
      0658 D480      CALL L0680 ; Lecture 2ème octet
      065A A0        MOV @R0,A ; Bits D7 à D0 du registre
     065B 98F8      ANL BUS,#11111000B ; DI, CK et CS forcés à zéro
      065D 18        INC R0 ; Adresse où stocker l'octet/le mot suivant
      065E 1D        INC R5 ; Registre suivant
     065F EC4C      DJNZ R4,L064C ; Lecture de tous les mots de 16 bits
      0661 1C        INC R4 ; R4 = 1
   0662 C8        DEC R0
      0663 C8        DEC R0 ; R0 = adresse du dernier mot reçu
      0664 CD        DEC R5 ; R5 = numéro dernier registre lu
  0665 83        RET

                ;
                 ; ********************************
                 ; Emission d'un ordre vers l'EEROM
                  ; R3 = ordre à émettre
                 ; ********************************

                ;
                L0666:

     0666 98F8      ANL BUS,#11111000B ; DI, CK et CS forcés à zéro
     0668 8804      ORL BUS,#00000100B ; CS à 1
  066A 00        NOP
  066B 00        NOP
     066C 8802      ORL BUS,#00000010B ; CK à 1
  066E 00        NOP
  066F 00        NOP
  0670 00        NOP
  0671 00        NOP
      0672 97        CLR C ; Bit de start dans Carry
   0673 A7        CPL C
      0674 D48B      CALL L068B ; Emission du bit de start
  0676 00        NOP
      0677 FB        MOV A,R3 ; Ordre à émettre contenu dans R3

                ;
                 ; ********************************
                 ; Ecriture d'un octet
                  ; A = octet à émettre/écrire
                 ; ********************************

                ;
                L0678:

      0678 BA08      MOV R2,#8 ; 8 bits à émettre
                L067A:

   067A F7        RLC A
      067B D48B      CALL L068B ; Bit à émettre dans Carry
     067D EA7A      DJNZ R2,L067A ; Emission des 8 bits
  067F 83        RET

                ;
                 ; ****************************
                 ; Lecture d'un octet
                  ; A = R3 = octet extrait
                 ; ****************************

                ;
                L0680:

      0680 BA08      MOV R2,#8 ; 8 bits à extraire
                L0682:

      0682 D49C      CALL L069C ; Extraction des bits
   0684 FB        MOV A,R3
      0685 F7        RLC A ; Sérialisation
   0686 AB        MOV R3,A
     0687 EA82      DJNZ R2,L0682 ; Extraction des 8 bits
   0689 FB        MOV A,R3
  068A 83        RET

                ;
                 ; ***********************
                 ; Ecriture d'un seul bit
                  ; Cy = bit à écrire
                 ; ***********************

                ;

                L068B:
     068B 98FD      ANL BUS,#11111101B ; CK à zéro
      068D B902      MOV R1,#2 ; Toute petite temporisation

                L068F:
   068F E98F      DJNZ R1,L068F
   0691 F697      JC L0697
     0693 98FE      ANL BUS,#11111110B ; DI à 0 si Carry à zéro
   0695 C499      JMP L0699

                ;
                L0697:

     0697 8801      ORL BUS,#00000001B ; DI à 1 si Carry = 1
                L0699:

     0699 8802      ORL BUS,#00000010B ; CK à 1
  069B 83        RET

                ;
                 ; **********************
                 ; Lecture d'un seul bit
                  ; Cy = bit extrait
                 ; **********************

                ;
                L069C:

     069C 98FD      ANL BUS,#11111101B ; CK à zéro
      069E B902      MOV R1,#2 ; Toute petite temporisation

                L06A0:
   06A0 E9A0      DJNZ R1,L06A0
   06A2 97        CLR C
      06A3 A7        CPL C ; Cy = 1, 
   06A4 09        IN A,P1
      06A5 32A8      JB1 L06A8 ; Cy = 1 si bit lu = 1 sinon Cy = 0
   06A7 A7        CPL C

                L06A8:
     06A8 8802      ORL BUS,#00000010B ; CK à 1
  06AA 83        RET

                ;
                 ; ********************************************
                 ; Temporisation d'écriture pour l'EEROM ~20 ms
                 ; ********************************************

                ;
                L06AB:

   06AB BA32      MOV R2,#50
                L06AD:

   06AD E9AD      DJNZ R1,L06AD
   06AF EAAD      DJNZ R2,L06AD
  06B1 83        RET

                ;
                 ; ************************************************************
                 ; ************************************************************
                 ; UART logicielle : Transmission en half-duplex, Format : 8N1
                 ; Vitesse de réception à 4800 Bds avec résonnateur de 10,4 MHz
                 ; ************************************************************
                 ; ************************************************************

                ;
                 ; ****************************************
                 ; Réception d'une série d'octets
                 ; P10 : entrée/sortie de l'UART logicielle
                 ; R0 : adresse zone de réceptrice
                 ; R4 : nombre d'octets attendus
                 ; A : dernier octet reçu
                 ; ****************************************

                ;
                L06B2:

      06B2 09        IN A,P1 ; Lecture du port P10
      06B3 12B2      JB0 L06B2 ; Attente état 0 : bit de start
      06B5 B924      MOV R1,#36 ; Temporisation de ½ moment

                L06B7:
   06B7 E9B7      DJNZ R1,L06B7
      06B9 BA08      MOV R2,#8 ; 8 bits à désérialiser

                L06BB:
      06BB B943      MOV R1,#67 ; Temporisation de 1 moment

                L06BD:
   06BD E9BD      DJNZ R1,L06BD
   06BF 09        IN A,P1
      06C0 67        RRC A ; Lecture du port P10
   06C1 F0        MOV A,@R0
      06C2 67        RRC A ; Insertion dans le registre récepteur
   06C3 A0        MOV @R0,A
   06C4 EABB      DJNZ R2,L06BB

                L06C6:
   06C6 09        IN A,P1
   06C7 37        CPL A
      06C8 12C6      JB0 L06C6 ; Attente état 1 : bit de stop
   06CA 18        INC R0
     06CB ECB2      DJNZ R4,L06B2 ; Réception octet suivant
   06CD 1C        INC R4
   06CE C8        DEC R0
      06CF F0        MOV A,@R0 ; Dernier octet reçu dans A
  06D0 83        RET

                ;
                 ; ****************************************
                 ; Emission d'une série d'octets
                 ; P10 : entrée/sortie de l'UART logicielle
                 ; R0 : adresse octets à émettre
                 ; R4 : nombre d'octets à émettre
                 ; A : dernier octet émis

                 ; ****************************************
                ;
                L06D1:

      06D1 B900      MOV R1,#0 ; Temporisation longue
                L06D3:

   06D3 E9D3      DJNZ R1,L06D3
                L06D5:

   06D5 E9D5      DJNZ R1,L06D5
      06D7 97        CLR C ; Bit de start dans Cy
      06D8 BA09      MOV R2,#9 ; Bit de start + 8 bits par octet
      06DA F0        MOV A,@R0 ; Octet à émettre
      06DB C4DE      JMP L06DE ; Emission du bit de start en premier

                L06DD:
   06DD 67        RRC A

                L06DE:
   06DE F6E4      JC L06E4
     06E0 99FE      ANL P1,#11111110B ; P10 à 0 si bit à émettre = 0
   06E2 C4E8      JMP L06E8

                L06E4:
     06E4 8901      ORL P1,#00000001B ; P10 à 1 si bit à émettre = 1
   06E6 C4E8      JMP L06E8

                L06E8:
      06E8 B942      MOV R1,#66 ; Temporisation pour 1 moment

                L06EA:
   06EA E9EA      DJNZ R1,L06EA
  06EC 00        NOP
     06ED EADD      DJNZ R2,L06DD ; Emission de tous les bits
     06EF 8901      ORL P1,#00000001B ; Emission du bit de stop
      06F1 18        INC R0 ; Octet à émettre suivant
   06F2 ECD1      DJNZ R4,L06D1
   06F4 C8        DEC R0
      06F5 F0        MOV A,@R0 ; Dernier octet émis dans A
   06F6 1C        INC R4
  06F7 83        RET

                ;
      0700           ORG 0700H ; Page 7

                ;
                 ; ********************************
                 ; Calcul du checksum sur la page 7
                 ; ********************************

                ;
                L0700:

      0700 FA        MOV A,R2 ; Adresse sur 8 bits dans la page
   0701 A3        MOVP A,@A
      0702 F4ED      CALL L07ED ; Calcul de checksum sur 16 bits
   0704 EA00      DJNZ R2,L0700
      0706 E4E3      JMP L07E3 ; Calcul terminé

                ;
                 ; ***********************************************************************
                 ; Test du clavier
                  ; Frappe des touches attendue dans l'ordre suivant : 
                  ; 1, 2, 3, 4, 5, 6, 7, 8, 9, 0, ENT et MEMO
                 ; Test terminé sur nouvelle frappe de la touche MEMO
                 ; ***********************************************************************

                ;
                L0708:

     0708 989F      ANL BUS,#10011111B ; Allumage LED jaune et LED verte
      070A BB0A      MOV R3,#10 ; 10 × ~20 ms

                L070C:
      070C D4AB      CALL L06AB ; Temporisation ~20 ms
   070E EB0C      DJNZ R3,L070C
     0710 8860      ORL BUS,#01100000B ; Extinction des deux LEDs
      0712 BC01      MOV R4,#1 ; Compteur de nombre de touches frappées

                L0714:
      0714 B458      CALL L0558 ; Attente frappe d'une touche sur le clavier
   0716 DC        XRL A,R4
      0717 9657      JNZ L0757 ; Chiffre frappé différent du compteur de frappe 
   0719 1C        INC R4
      071A FC        MOV A,R4 ; Compteur de frappes incrémenté
   071B D30A      XRL A,#10
      071D 9614      JNZ L0714 ; Tant que la 10ème touche n'a pas été frappée
      071F B458      CALL L0558 ; Attente frappe de la 10ème touche sur le clavier
   0721 D300      XRL A,#0 
      0723 9657      JNZ L0757 ; Cette touche doit être celle du chiffre zéro
      0725 B458      CALL L0558 ; Attente frappe de la 11ème touche sur le clavier
   0727 D310      XRL A,#ENT
      0729 9657      JNZ L0757 ; Cette touche doit être la touche "ENT"
      072B B458      CALL L0558 ; Attente frappe de la 12ème touche sur le clavier
      072D D320      XRL A,#MEMO ; Cette touche doit être la touche "MEMO" jaune
   072F 9657      JNZ L0757
     0731 98DF      ANL BUS,#11011111B ; Allumage LED verte si test OK

                L0733:
      0733 B458      CALL L0558 ; Attente nouvelle frappe de la touche MEMO
   0735 D320      XRL A,#MEMO
      0737 9633      JNZ L0733 ; Fin de test si touche MEMO frappée
  0739 83        RET

                ;
                 ; ***********************************
                 ; Test d'écriture dans l'EEROM
                 ; Effectué uniquement sur registre 15
                 ; ***********************************

                ;
                L073A:

     073A 8860      ORL BUS,#01100000B ; Extinction des deux LEDs

     073C B826      MOV R0,#RAM026 ; Adresse du mot à écrire
     073E B055      MOV @R0,#01010101B ; Motif à écrire
      0740 BD0F      MOV R5,#15 ; Registre 15 de l'EEROM
      0742 BC01      MOV R4,#1 ; Ecriture d'un seul mot
      0744 D41A      CALL L061A ; Ecriture du motif dans le registre 15 de l'EEROM
      0746 D44C      CALL L064C ; Lecture du registre 15 de l'EEROM
   0748 F0        MOV A,@R0
     0749 D355      XRL A,#01010101B ; Comparaison entre le motif lu et le motif écrit
   074B 9657      JNZ L0757
     074D 98DF      ANL BUS,#11011111B ; Allumage LED verte si test OK

                L074F:
      074F B458      CALL L0558 ; Attente frappe de la touche MEMO
   0751 D320      XRL A,#MEMO
      0753 964F      JNZ L074F ; Fin de test si touche MEMO frappée
      0755 0410      JMP L0010 ; Réinitialisation de l'appareil

                ;
                L0757:

     0757 98BF      ANL BUS,#10111111B ; Allumage LED jaune si erreur
                L0759:

      0759 E459      JMP L0759 ; Boucle jusqu'au prochain reset
                ;

                 ;****************************************************************************
                 ; Conversion du code périodique de 10 chiffres en nombre binaire de 32 bits
                 ; Les 10 chiffres du code périodique sont stockés en RAM à partir de RAM05F
                 ; Le nombre binaire est retourné en RAM020, octet le plus significatif en 1er
                 ; *************************************************************************** 

                L075B:
     075B B820      MOV R0,#RAM020 ; Adresse zone RAM pour la conversion binaire
      075D BA04      MOV R2,#4 ; Nombre d'octets à initialiser
      075F 54EF      CALL L02EF ; Initialisation
     0761 B85F      MOV R0,#RAM05F ; Adresse des 10 chiffres du code périodique
     0763 BAB5      MOV R2,#LOW L07B5 ; Table des puissances de 10

                L0765:
      0765 F0        MOV A,@R0 ; Lecture d'un chiffre
   0766 AC        MOV R4,A
      0767 C66F      JZ L076F ; Pas de calcul pour le chiffre zéro

                L0769:
   0769 FA        MOV A,R2
      076A A3        MOVP A,@A ; 1ère donnée, la plus significative
   076B F4B0      CALL L07B0
     076D EC69      DJNZ R4,L0769 ; Additionnée autant de fois que la valeur du chiffre

                L076F:
      076F 1A        INC R2 ; 2ème donnée de la puissance de dix
      0770 F0        MOV A,@R0 ; Relecture du même chiffre
   0771 AC        MOV R4,A
      0772 C67A      JZ L077A ; Pas de calcul pour le chiffre zéro

                L0774:
   0774 FA        MOV A,R2
      0775 A3        MOVP A,@A ; 2ème donnée, intermédiaire haute
      0776 F4A8      CALL L07A8 ; 
     0778 EC74      DJNZ R4,L0774 ; Additionnée autant de fois que la valeur du chiffre

                L077A:
      077A 1A        INC R2 ; 3ème donnée de la puissance de dix
      077B F0        MOV A,@R0 ; Relecture du même chiffre
   077C AC        MOV R4,A
      077D C685      JZ L0785 ; Pas de calcul pour le chiffre zéro

                L077F:
   077F FA        MOV A,R2
      0780 A3        MOVP A,@A ; 3ème donnée, intermédiaire basse
   0781 F4A0      CALL L07A0
     0783 EC7F      DJNZ R4,L077F ; Additionnée autant de fois que la valeur du chiffre

                L0785:
      0785 1A        INC R2 ; 4ème et dernière donnée de la puissance de dix
      0786 F0        MOV A,@R0 ; Relecture du même chiffre
   0787 AC        MOV R4,A
      0788 C690      JZ L0790 ; Pas de calcul pour le chiffre zéro

                L078A:
   078A FA        MOV A,R2
      078B A3        MOVP A,@A ; 4ème donnée, la moins significative
   078C F498      CALL L0798
     078E EC8A      DJNZ R4,L078A ; Additionnée autant de fois que la valeur du chiffre

                L0790:
      0790 1A        INC R2 ; Réitération avec la puissance de 10 en-dessous
      0791 18        INC R0 ; Chiffres suivant
   0792 F8        MOV A,R0
     0793 D369      XRL A,#RAM05F + 10 ; Zone limitée à 10 chiffres
      0795 9665      JNZ L0765 ; Bouclage jusqu'au chiffre des unités
  0797 83        RET

                ;
                 ; Addition donnée inférieure de la puissance de 10

                ;
                L0798:

   0798 B923      MOV R1,#RAM020 + 3
   079A 61        ADD A,@R1
   079B A1        MOV @R1,A
      079C E6B4      JNC L07B4 ; Pas de repport si Cy = 0
   079E 2301      MOV A,#1

                ;
                 ; Addition donnée intermédiaire basse

                ;
                L07A0:

   07A0 B922      MOV R1,#RAM020 + 2
   07A2 61        ADD A,@R1
   07A3 A1        MOV @R1,A

      07A4 E6B4      JNC L07B4 ; Pas de repport si Cy = 0
   07A6 2301      MOV A,#1

                ;
                 ; Addition donnée intermédiaire haute

                 ;
                L07A8:

   07A8 B921      MOV R1,#RAM020 + 1
   07AA 61        ADD A,@R1
   07AB A1        MOV @R1,A
      07AC E6B4      JNC L07B4 ; Pas de repport si Cy = 0
   07AE 2301      MOV A,#1

                ;
                 ; Addition donnée supérieure de la puissance de 10

                ;
                L07B0:

   07B0 B920      MOV R1,#RAM020
   07B2 61        ADD A,@R1
   07B3 A1        MOV @R1,A

                L07B4:
  07B4 83        RET

                ;
                 ; ************************************************
                 ; Table des puissances de 10, données sur 4 octets
                 ; ************************************************

                ;
                L07B5:

    07B5 3B 9A CA  DB 3BH, 9AH, 0CAH, 0 ; 10^9
 07B8 00 

    07B9 05 F5 E1  DB 5, 0F5H, 0E1H, 0 ; 10^8
 07BC 00 

    07BD 00 98 96  DB 0, 98H, 96H, 80H ; 10^7
 07C0 80 

    07C1 00 0F 42  DB 0, 0FH, 42H, 40H ; 10^6
 07C4 40 

     07C5 00 01 86  DB 0, 1, 86H, 0A0H ; 10^5
 07C8 A0 

     07C9 00 00 27  DB 0, 0, 27H, 10H ; 10^4
 07CC 10 

     07CD 00 00 03  DB 0, 0, 3,0E8H ; 10^3
 07D0 E8 

     07D1 00 00 00  DB 0, 0, 0, 100 ; 10^2
 07D4 64 

     07D5 00 00 00  DB 0, 0, 0, 10 ; 10^1
 07D8 0A 

     07D9 00 00 00  DB 0, 0, 0, 1 ; 10^0
 07DC 01 
                ;

                 ; **********************************************************
                 ; Calcul du checksum de la ROM interne initié par P27 à zéro
                 ; **********************************************************

                ;
                L07DD:

   07DD 27        CLR A
      07DE A8        MOV R0,A ; Reset octet de poids faible du checksum
      07DF A9        MOV R1,A ; Reset octet de poids fort du checksum
      07E0 AA        MOV R2,A ; Adresse de départ dans la page
      07E1 0408      JMP L0008 ; Départ du calcul sur la page zéro

                ;
                 ; *****************************
                 ; Sortie du checksum en 2 temps
                 ; *****************************

                ;
                L07E3:

      07E3 F8        MOV A,R0 ; Sortie de l'octet de poids faible en premier
   07E4 02        OUTL BUS,A
     07E5 997F      ANL P1,#01111111B ; P17 à zéro
      07E7 F9        MOV A,R1 ; Sortie de l'octet de poids fort en second
   07E8 02        OUTL BUS,A
     07E9 993F      ANL P1,#00111111B ; P17 et P16 à zéro

                L07EB:
      07EB E4EB      JMP L07EB ; Boucle indéfiniment jusqu'au prochain reset

                ;
                 ; ***********************************
                 ; Ajout au checksum d'un nouvel octet
                 ; ***********************************

                ;
                L07ED:

   07ED 68        ADD A,R0
      07EE A8        MOV R0,A ; Octet de poids faible
   07EF 27        CLR A
      07F0 79        ADDC A,R1 ; Report éventuel sur octet de poids fort
   07F1 A9        MOV R1,A
  07F2 83        RET

                ;
   0000           END L0000

25



RT86.LST
The Cybernetic Micro Systems 8048 Family Assembler, Version 3.00   Page  26

 
 ;%T   Symbol Name            Type Value
 
 DEFAUT. . . . . . . . . . . . . I 29CB
 ENT . . . . . . . . . . . . . . I 0010
 ERASE . . . . . . . . . . . . . I 00C0
 EWDS. . . . . . . . . . . . . . I 0000
 EWEN. . . . . . . . . . . . . . I 0030
 L0000 . . . . . . . . . . . . . L 0000
 L0008 . . . . . . . . . . . . . L 0008
 L0010 . . . . . . . . . . . . . L 0010
 L0015 . . . . . . . . . . . . . L 0015
 L001B . . . . . . . . . . . . . L 001B
 L001F . . . . . . . . . . . . . L 001F
 L0027 . . . . . . . . . . . . . L 0027
 L004F . . . . . . . . . . . . . L 004F
 L0059 . . . . . . . . . . . . . L 0059
 L005D . . . . . . . . . . . . . L 005D
 L0063 . . . . . . . . . . . . . L 0063
 L0065 . . . . . . . . . . . . . L 0065
 L007B . . . . . . . . . . . . . L 007B
 L0081 . . . . . . . . . . . . . L 0081
 L0083 . . . . . . . . . . . . . L 0083
 L008F . . . . . . . . . . . . . L 008F
 L0097 . . . . . . . . . . . . . L 0097
 L009D . . . . . . . . . . . . . L 009D
 L00A5 . . . . . . . . . . . . . L 00A5
 L00A7 . . . . . . . . . . . . . L 00A7
 L00AB . . . . . . . . . . . . . L 00AB
 L00B1 . . . . . . . . . . . . . L 00B1
 L00B3 . . . . . . . . . . . . . L 00B3
 L00C1 . . . . . . . . . . . . . L 00C1
 L00CB . . . . . . . . . . . . . L 00CB
 L00D3 . . . . . . . . . . . . . L 00D3
 L0100 . . . . . . . . . . . . . L 0100
 L0108 . . . . . . . . . . . . . L 0108
 L012B . . . . . . . . . . . . . L 012B
 L0139 . . . . . . . . . . . . . L 0139
 L0142 . . . . . . . . . . . . . L 0142
 L014B . . . . . . . . . . . . . L 014B
 L014D . . . . . . . . . . . . . L 014D
 L0158 . . . . . . . . . . . . . L 0158
 L015A . . . . . . . . . . . . . L 015A
 L0168 . . . . . . . . . . . . . L 0168
 L0169 . . . . . . . . . . . . . L 0169
 L016B . . . . . . . . . . . . . L 016B
 L017D . . . . . . . . . . . . . L 017D
 L0186 . . . . . . . . . . . . . L 0186
 L018B . . . . . . . . . . . . . L 018B
 L0199 . . . . . . . . . . . . . L 0199
 L019D . . . . . . . . . . . . . L 019D
 L01A0 . . . . . . . . . . . . . L 01A0
 L01B4 . . . . . . . . . . . . . L 01B4
 L01B8 . . . . . . . . . . . . . L 01B8
 L01B9 . . . . . . . . . . . . . L 01B9
 L01D9 . . . . . . . . . . . . . L 01D9
 L01DF . . . . . . . . . . . . . L 01DF
 L01E9 . . . . . . . . . . . . . L 01E9
 L0200 . . . . . . . . . . . . . L 0200
 L0208 . . . . . . . . . . . . . L 0208
 L0210 . . . . . . . . . . . . . L 0210
 L0219 . . . . . . . . . . . . . L 0219
 L0221 . . . . . . . . . . . . . L 0221
 L0223 . . . . . . . . . . . . . L 0223
 L0229 . . . . . . . . . . . . . L 0229
 L022D . . . . . . . . . . . . . L 022D
 L023D . . . . . . . . . . . . . L 023D
 L023F . . . . . . . . . . . . . L 023F
 L0248 . . . . . . . . . . . . . L 0248
 L024D . . . . . . . . . . . . . L 024D
 L024F . . . . . . . . . . . . . L 024F
 L0257 . . . . . . . . . . . . . L 0257
 L025C . . . . . . . . . . . . . L 025C
 L0265 . . . . . . . . . . . . . L 0265
 L026E . . . . . . . . . . . . . L 026E
 L0270 . . . . . . . . . . . . . L 0270
 L027A . . . . . . . . . . . . . L 027A
 L027C . . . . . . . . . . . . . L 027C
 L0284 . . . . . . . . . . . . . L 0284
 L028A . . . . . . . . . . . . . L 028A
 L0290 . . . . . . . . . . . . . L 0290
 L0296 . . . . . . . . . . . . . L 0296
 L0299 . . . . . . . . . . . . . L 0299
 L02A5 . . . . . . . . . . . . . L 02A5
 L02A6 . . . . . . . . . . . . . L 02A6
 L02A8 . . . . . . . . . . . . . L 02A8
 L02AA . . . . . . . . . . . . . L 02AA
 L02AC . . . . . . . . . . . . . L 02AC
 L02AE . . . . . . . . . . . . . L 02AE
 L02B4 . . . . . . . . . . . . . L 02B4
 L02BB . . . . . . . . . . . . . L 02BB
 L02CA . . . . . . . . . . . . . L 02CA
 L02D6 . . . . . . . . . . . . . L 02D6
 L02E3 . . . . . . . . . . . . . L 02E3

 L02E4 . . . . . . . . . . . . . L 02E4
 L02EC . . . . . . . . . . . . . L 02EC
 L02EF . . . . . . . . . . . . . L 02EF
 L02F0 . . . . . . . . . . . . . L 02F0
 L0300 . . . . . . . . . . . . . L 0300
 L0308 . . . . . . . . . . . . . L 0308
 L0313 . . . . . . . . . . . . . L 0313
 L0320 . . . . . . . . . . . . . L 0320
 L0326 . . . . . . . . . . . . . L 0326
 L0335 . . . . . . . . . . . . . L 0335
 L0337 . . . . . . . . . . . . . L 0337
 L0339 . . . . . . . . . . . . . L 0339
 L0345 . . . . . . . . . . . . . L 0345
 L0351 . . . . . . . . . . . . . L 0351
 L0357 . . . . . . . . . . . . . L 0357
 L035D . . . . . . . . . . . . . L 035D
 L0370 . . . . . . . . . . . . . L 0370
 L0382 . . . . . . . . . . . . . L 0382
 L0388 . . . . . . . . . . . . . L 0388
 L0392 . . . . . . . . . . . . . L 0392
 L039A . . . . . . . . . . . . . L 039A
 L039E . . . . . . . . . . . . . L 039E
 L03AC . . . . . . . . . . . . . L 03AC
 L03AE . . . . . . . . . . . . . L 03AE
 L03B4 . . . . . . . . . . . . . L 03B4
 L03B9 . . . . . . . . . . . . . L 03B9
 L03C3 . . . . . . . . . . . . . L 03C3
 L03D0 . . . . . . . . . . . . . L 03D0
 L03D2 . . . . . . . . . . . . . L 03D2
 L03DD . . . . . . . . . . . . . L 03DD
 L03ED . . . . . . . . . . . . . L 03ED
 L0400 . . . . . . . . . . . . . L 0400
 L0408 . . . . . . . . . . . . . L 0408
 L0422 . . . . . . . . . . . . . L 0422
 L0427 . . . . . . . . . . . . . L 0427
 L0439 . . . . . . . . . . . . . L 0439
 L043E . . . . . . . . . . . . . L 043E
 L044A . . . . . . . . . . . . . L 044A
 L0455 . . . . . . . . . . . . . L 0455
 L045D . . . . . . . . . . . . . L 045D
 L046B . . . . . . . . . . . . . L 046B
 L046C . . . . . . . . . . . . . L 046C
 L0478 . . . . . . . . . . . . . L 0478
 L047D . . . . . . . . . . . . . L 047D
 L0486 . . . . . . . . . . . . . L 0486
 L0492 . . . . . . . . . . . . . L 0492
 L0494 . . . . . . . . . . . . . L 0494
 L049A . . . . . . . . . . . . . L 049A
 L04A3 . . . . . . . . . . . . . L 04A3
 L04AF . . . . . . . . . . . . . L 04AF
 L04B8 . . . . . . . . . . . . . L 04B8
 L04D7 . . . . . . . . . . . . . L 04D7
 L04E3 . . . . . . . . . . . . . L 04E3
 L04E4 . . . . . . . . . . . . . L 04E4
 L04EB . . . . . . . . . . . . . L 04EB
 L04EF . . . . . . . . . . . . . L 04EF
 L04F5 . . . . . . . . . . . . . L 04F5
 L0500 . . . . . . . . . . . . . L 0500
 L0508 . . . . . . . . . . . . . L 0508
 L0512 . . . . . . . . . . . . . L 0512
 L051E . . . . . . . . . . . . . L 051E
 L053D . . . . . . . . . . . . . L 053D
 L053E . . . . . . . . . . . . . L 053E
 L0541 . . . . . . . . . . . . . L 0541
 L054A . . . . . . . . . . . . . L 054A
 L0557 . . . . . . . . . . . . . L 0557
 L0558 . . . . . . . . . . . . . L 0558
 L055C . . . . . . . . . . . . . L 055C
 L055E . . . . . . . . . . . . . L 055E
 L056A . . . . . . . . . . . . . L 056A
 L0570 . . . . . . . . . . . . . L 0570
 L0576 . . . . . . . . . . . . . L 0576
 L0579 . . . . . . . . . . . . . L 0579
 L058B . . . . . . . . . . . . . L 058B
 L0595 . . . . . . . . . . . . . L 0595
 L0596 . . . . . . . . . . . . . L 0596
 L05A0 . . . . . . . . . . . . . L 05A0
 L05AD . . . . . . . . . . . . . L 05AD
 L05AF . . . . . . . . . . . . . L 05AF
 L05B0 . . . . . . . . . . . . . L 05B0
 L05BC . . . . . . . . . . . . . L 05BC
 L05C3 . . . . . . . . . . . . . L 05C3
 L05DB . . . . . . . . . . . . . L 05DB
 L05E4 . . . . . . . . . . . . . L 05E4
 L05E6 . . . . . . . . . . . . . L 05E6
 L05EA . . . . . . . . . . . . . L 05EA
 L0600 . . . . . . . . . . . . . L 0600
 L0608 . . . . . . . . . . . . . L 0608
 L0611 . . . . . . . . . . . . . L 0611
 L061A . . . . . . . . . . . . . L 061A
 L0620 . . . . . . . . . . . . . L 0620
 L064C . . . . . . . . . . . . . L 064C
 L0666 . . . . . . . . . . . . . L 0666
 L0678 . . . . . . . . . . . . . L 0678

27



RT86.LST
The Cybernetic Micro Systems 8048 Family Assembler, Version 3.00   Page  28

 L067A . . . . . . . . . . . . . L 067A
 L0680 . . . . . . . . . . . . . L 0680
 L0682 . . . . . . . . . . . . . L 0682
 L068B . . . . . . . . . . . . . L 068B
 L068F . . . . . . . . . . . . . L 068F
 L0697 . . . . . . . . . . . . . L 0697
 L0699 . . . . . . . . . . . . . L 0699
 L069C . . . . . . . . . . . . . L 069C
 L06A0 . . . . . . . . . . . . . L 06A0
 L06A8 . . . . . . . . . . . . . L 06A8
 L06AB . . . . . . . . . . . . . L 06AB
 L06AD . . . . . . . . . . . . . L 06AD
 L06B2 . . . . . . . . . . . . . L 06B2
 L06B7 . . . . . . . . . . . . . L 06B7
 L06BB . . . . . . . . . . . . . L 06BB
 L06BD . . . . . . . . . . . . . L 06BD
 L06C6 . . . . . . . . . . . . . L 06C6
 L06D1 . . . . . . . . . . . . . L 06D1
 L06D3 . . . . . . . . . . . . . L 06D3
 L06D5 . . . . . . . . . . . . . L 06D5
 L06DD . . . . . . . . . . . . . L 06DD
 L06DE . . . . . . . . . . . . . L 06DE
 L06E4 . . . . . . . . . . . . . L 06E4
 L06E8 . . . . . . . . . . . . . L 06E8
 L06EA . . . . . . . . . . . . . L 06EA
 L0700 . . . . . . . . . . . . . L 0700
 L0708 . . . . . . . . . . . . . L 0708
 L070C . . . . . . . . . . . . . L 070C
 L0714 . . . . . . . . . . . . . L 0714
 L0733 . . . . . . . . . . . . . L 0733
 L073A . . . . . . . . . . . . . L 073A
 L074F . . . . . . . . . . . . . L 074F
 L0757 . . . . . . . . . . . . . L 0757
 L0759 . . . . . . . . . . . . . L 0759
 L075B . . . . . . . . . . . . . L 075B
 L0765 . . . . . . . . . . . . . L 0765
 L0769 . . . . . . . . . . . . . L 0769
 L076F . . . . . . . . . . . . . L 076F
 L0774 . . . . . . . . . . . . . L 0774
 L077A . . . . . . . . . . . . . L 077A
 L077F . . . . . . . . . . . . . L 077F
 L0785 . . . . . . . . . . . . . L 0785
 L078A . . . . . . . . . . . . . L 078A
 L0790 . . . . . . . . . . . . . L 0790
 L0798 . . . . . . . . . . . . . L 0798
 L07A0 . . . . . . . . . . . . . L 07A0
 L07A8 . . . . . . . . . . . . . L 07A8
 L07B0 . . . . . . . . . . . . . L 07B0
 L07B4 . . . . . . . . . . . . . L 07B4
 L07B5 . . . . . . . . . . . . . L 07B5
 L07DD . . . . . . . . . . . . . L 07DD
 L07E3 . . . . . . . . . . . . . L 07E3
 L07EB . . . . . . . . . . . . . L 07EB
 L07ED . . . . . . . . . . . . . L 07ED
 MEMO. . . . . . . . . . . . . . I 0020
 R0000 . . . . . . . . . . . . . I 00AF
 R0902 . . . . . . . . . . . . . I 009F
 R1804 . . . . . . . . . . . . . I 00FF
 RAM020. . . . . . . . . . . . . I 0020
 RAM026. . . . . . . . . . . . . I 0026
 RAM029. . . . . . . . . . . . . I 0029
 RAM02A. . . . . . . . . . . . . I 002A
 RAM02B. . . . . . . . . . . . . I 002B
 RAM02C. . . . . . . . . . . . . I 002C
 RAM02F. . . . . . . . . . . . . I 002F
 RAM030. . . . . . . . . . . . . I 0030
 RAM048. . . . . . . . . . . . . I 0048
 RAM04E. . . . . . . . . . . . . I 004E
 RAM04F. . . . . . . . . . . . . I 004F
 RAM050. . . . . . . . . . . . . I 0050
 RAM052. . . . . . . . . . . . . I 0052
 RAM05F. . . . . . . . . . . . . I 005F
 RAM06A. . . . . . . . . . . . . I 006A
 RAM070. . . . . . . . . . . . . I 0070
 READ. . . . . . . . . . . . . . I 0080
 SESAME. . . . . . . . . . . . . I A72C
 TSIZE . . . . . . . . . . . . . I 000C
 WRITE . . . . . . . . . . . . . I 0040

